
1.  什麼是 GeoGebra? 

GeoGebra 是一套結合幾何、代數及微積分的動態數學軟體。目前是由 Markus 
Hohenwarter 及一個國際程式設計團隊所共同開發，目的在教導與學習學校數學。 

1.1.  數學物件的多重視區 

GeoGebra 提供數學物件的三種不同視區：幾何視區、代數視區和試算表視區。這些

視區能夠將數學物件以三種不同形式的表徵呈現：圖形化 (例：點、函數圖形)、 代數

化 (例：點座標、方程式)與試算表中表列呈現。因此，不論物件一開始如何被產生出

來，同一個物件的所有表徵會動態連結以及同步調整其他表徵。 
 

 

1.1.1.  幾何視區 

可以藉由工具列中所提供的構圖工具，在幾何視區使用滑鼠進行幾何構圖。從工具列

中選擇其中一種構圖工具，可藉由工具列說明 (在工具列後方)來得知如何使用所選擇

的工具。任何在幾何視區所產生的物件，在代數視區都會有一個代數表徵。 
 
註: 可以使用滑鼠拖曳來移動幾何視區的物件。代數視區的代數表徵也會同時動態更

新。 
 
工具列中的每一個圖示代表一個工具盒，包含一些相似的構圖工具。可以按下圖示右

下角的小箭頭來打開工具盒。 

代數視區 

幾何視區 

試算表視區 

指令列 

工具列 



7 
 

 
提示: 構圖工具是依照產生物件的本質來分類。 可以在點工具盒 (預設圖示 ) 及幾何

變換工具盒(預設圖示 )產生不同類型的點。 

1.1.2.  代數視區 

可以在 GeoGebra 的指令列中 直接輸入代數表示式。 在按下 Enter-鍵後，所輸入的

代數表示式會在代數視區出現，同時在幾何視區出現圖形表徵。 
 
在代數視區中，數學物件被歸類成自變物件與應變物件。如果產生一個新物件而未使

用任何已存在的物件，則被歸類為自變物件。若新產生的物件是使用其他以存在的物

件產生，則歸類為應變物件。 
提示: 假如想要隱藏在代數視區的代數表徵，可以將物件指定為輔助物件：在代數視區

中，對物件按右鍵(MacOS: Ctrl-click) 並在內容選單中選擇 ‘屬性’ 。在屬性對話框中

的標籤 ‘一般’ 將物件指定為 ‘輔助物件’。根據預設，輔助物件並不會出現在代數視

區，但是可以在檢視選單中選擇 ‘輔助物件’。 
 
可以 修改代數視區的物件 ：先確認已選取  移動 工具，然後雙擊代數視區的自變物

件前。可以在文字方塊中直接編輯物件的代數表徵。在按下 Enter-鍵後，物件的圖形

表徵也會自動改變。 
若雙擊代數視區中的應變物件時，可以在出現的對話框中重新定義 物件。 
 
GeoGebra 也提供許多指令。可以在指令列的右方按下‘Command’按鈕來開啟指令列

表。在選單中選擇一個指令後 (或直接在指令列中輸入指令名稱) 可以按下 F1-鍵 取得

指令的語法結構以及所需的引數等資訊。 

1.1.3.  試算表視區 

在 GeoGebra 的 試算表視區中，每一個表格都有指定名稱 用來指定表格的位置。

例，在 A 欄 第 1 列的表格名稱為 A1。  
註: 當表格名稱用在表達式或指令中，表格名稱代表的是對應表格內的內容。 
 
在試算表表格中，不但可以輸入數值，也可以輸入所有 GeoGebra 所支援的數學物件

(例：點座標, 函數, 指令)。在試算表中所輸入數學物件，GeoGebra 會立即顯示其圖

形的表徵於幾何視區，而且，物件名稱會對應試算表格中的位置。(例：A5, C1)。 
 
註: 根據預設，試算表中的物件在代數視區中被歸類為輔助物件。可以在檢視選單中選

擇‘輔助物件’來顯示或隱藏輔助物件。 

1.2.  GeoGebra 作為數學學習與教學工具 

1.2.1.  自訂使用者介面 

GeoGebra 的使用者介面可以使用檢視選單來自訂。例，可以藉由取消檢視選單中的

對應選項來隱藏各個介面 (例：代數視區, 試算表視區, 或指令列)。 

自訂幾何視區  

可以顯示或隱藏幾何視區中的物件。使用工具  顯示/隱藏 物件 或 內容選單 來改變

物件顯示與否。在代數視區中， 每個物件左方的圖示顯示物件目前的顯示狀態 (  ‘顯
示’ 或  ‘隱藏’).  
註: 亦可以使用工具  顯示或隱藏群組物件之選項 來顯示或隱藏一個或多個物件。 
 
為了調整幾何視區的可視部分，可以透過工具  移動幾何視區 來移動幾何視區的背

景並使用下列方式縮放： 
 可以使用工具  放大 和  縮小 來縮放幾何視區。  

註: 所按下的位置會當作縮放中心。 
 可以使用滑鼠上的滾輪來縮放幾何視區 
 可以使用 快捷鍵 來放大 (Ctrl +) 和縮小 (Ctrl -)。 
 在幾何視區的空白處按下右鍵(MacOS: Ctrl - click)，內容選單也能讓做‘縮放’。 
 可以在幾何視區的空白處按右鍵 (MacOS: Cmd - click) 來指定縮放區域 並拖曳

滑鼠至要縮放區域的對頂角。當完成縮放區域後，放開滑鼠，將會自動調整幾

何視區的空間。 
 
可以使用檢視選單來顯示或隱藏幾何視區中的座標軸和格線。 
註: 另一個顯示或隱藏座標軸與格線的方式是在幾何視區中按右鍵 (MacOS: Ctrl-click) 
並從內容選單中選擇對應的選項  ‘座標軸’ 或  ‘格線’ 。 

自訂座標軸與格線 

可使用幾何視區中的屬性對話框來自訂座標軸與格線。在幾何視區中按下右鍵 
(MacOS: Ctrl-click) ，選擇幾何視區中出現的內容選單中的‘屬性’開啟對話視窗。 

 切換 ’座標軸’中，可以改變線的樣式和座標軸的單位與座標軸的範圍大小。藉

由選擇標籤‘x 軸’ 或 ‘y 軸’來自訂各個座標軸。也可以個別改變座標軸的比例大

小或顯示與否。 
 切換‘格線’中，可以改變顏色、格線的樣式及格線範圍大小，此外，也可將格線

設定為 ‘等距線’。 
 

註: 在任何模式下都可以藉由按住 Shift-鍵 (PC: Ctrl-鍵也可) 來拖曳座標軸以改變座標

軸的比例大小。 
 
註: 幾何視區中的屬性對話框與物件的 屬性對話框 是不同的。 

自訂工具列 

工具列的自訂可以藉由工具選單中選擇 ‘自訂工具列…’。 從出現的對話視窗左手邊選

擇想要從 GeoGebra 工具列中移除的工具或工具盒來移除工具列中的工具/工具盒。  
 
註: 可以藉由按下對話視窗左下角的‘回復預設工具列’按鈕來回復預設工具列。 

1.2.2.  改變物件屬性 

屬性對話框可以讓修改物件的屬性(例：顏色、線段樣式、顯示與否).  
 
可以使用下列幾種方式開啟屬性對話框： 

 在物件上按右鍵(MacOS: Ctrl - click) 並從出現的內容選單中選擇  ‘屬性…’。 
 從編輯選單中選擇  ‘屬性’。 
 選擇  移動 工具雙擊幾何視區中的物件。在 重新定義 對話框中，按下 ‘屬

性…’按鈕。 
 
在屬性對話框中，物件會依照形式歸類(例：點、線、圓) 在左手邊，有助於處理大量

的物件更為簡易。可以在串列中選擇一個或多個物件來改變屬性。 
註: 可以藉由點擊串列中的其中一個標題(例：‘點’)來選擇某一類型的所有物件，因此

可以快速地修改此類型的所有物件。 
 
可以使用右手邊的標籤(例：‘一般’, ‘顏色’, ‘樣式’, ‘進接’)來修改選擇物件的屬性。 
註: 標籤會依據串列中所選擇的物件而有所不同。 
 
當修改完物件屬性時，可以關閉屬性對話框。 

1.2.3.  使用內容選單 

內容選單提供一個快速的方式來改變物件的行為或進一步的屬性。在物件上按右鍵 
(MacOS: Ctrl-click) 來開啟內容選單。例如，可以改變物件的代數表示式(例：極座標

或直角座標, 隱性或顯性方程式?) 以及直接使用的功能，像是  重新命名,  刪除,  
顯示移動蹤跡, 開始滑動, 或  複製到指令列.  
 
註: 若在幾何視區中，對一個點開啟內容選單，會看到一個選項  ‘記錄蹤跡至試算表’  
(只有在開啟試算表視區才有). 一旦選取, 當被選取的點移動時，此功能將會幫記錄點

座標至試算表視區中。 
 
選擇內容選單中的  屬性… 來開啟 屬性對話框, 可以改變物件的屬性 (例：顏色, 大
小, 線寬, 線的樣式, 填色). 

1.3.  GeoGebra作為展示工具 

1.3.1.  使用導播列 

GeoGebra 提供導播列並允許透過此功能導播 GeoGebra 檔案的每個構圖步驟。於檢

視選單中選取‘構圖步驟的導播列’項目將顯示導播列於幾何視區的底部。 
 
導播列提供導播的按鈕並顯示其為第幾個步驟建構而成(例如：2/7 表示總共有七個步

驟，此為其中的第二個步驟。)： 
  按鈕：回到第一步驟。 
  按鈕：回到前一個步驟 
  按鈕：前進到下一個步驟。 
  按鈕：前進到最後一個步驟。 
  播放：自動的逐步播放構圖步驟。 

註：可以在播放按鈕 右邊的工作區塊中改變自動撥放的速度。 
  暫停：停止自動撥放。  

註：這個按鈕僅會顯示於按下播放按鈕之後。 
  按鈕： 這個按鈕會顯示 構圖案本。 

1.3.2.  使用構圖案本  

構圖案本介面的開啟可以從檢視選單中選取‘構圖案本’選項。它會以表格呈現所有

的構圖過程。藉由構圖案本對話框底部的導播列，這個功能可以讓逐步重作已經完成

的構圖。  

導播與修改構圖案本 

在構圖案本中，可以利用鍵盤進行導播： 
 使用鍵盤↑-鍵可以移動到前一步驟。 
 使用鍵盤↓-鍵可以移動到下一步驟。 
 使用鍵盤Home-鍵可以移動到構圖案本的第一步。 
 使用鍵盤End-鍵可以移動到構圖案本的最後一步。 
 使用鍵盤Delete -鍵可以刪除所選擇的步驟。 

註：這個動作可能會影響其他相依的物件或步驟。 
 
在構圖案本中，也可以使用滑鼠進行導播： 

 雙擊滑鼠左鍵選取想要選擇的步驟。 
 在標題欄上雙擊滑鼠左鍵可回到構圖起始狀態。 
 想要移動構圖步驟到其他位置可以利用滑鼠拖曳該列。 

註：這個動作不一定可以執行，若有相依的步驟則不一定可以移動。 
 點擊滑鼠右鍵於任一列可開啟此構圖步驟物件的內容選單。 

註：可以在任意位置插入構圖步驟：選取想要插入步驟的前一步驟，關閉構圖案本視

窗並開始建立新的物件，這新的構圖步驟將會被插入構圖案本已選取位置的下方。 
 
使用「暫停點」欄位(構圖案本視窗的檢視選單中)，可定義特定的構圖步驟為’暫停

點’，這個功能可以群組許多物件。當用導播列來導播構圖，物件的群組會同時出現。 
 
註：於構圖案本視窗的檢視選單中可切換構圖案本的各個欄位是否顯示。  

構圖案本以網頁輸出 

GeoGebra 可以將構圖案本輸出成網頁。首先必須從檢視選單中開啟構圖案本，然後

可在構圖案本視窗的檔案選單中點擊’網頁輸出’的選項。 
 
在構圖案本的輸出視窗中, 可輸入’標題’, ‘作者’和’日期’, 以及是否要包含代數視區與幾

何視區的圖像。此外，也能夠勾選’色彩的構圖案本’ ，意即每一個構圖步驟與顯示的

物件將會有相同的顏色顯示。 
 
註: 輸出的 HTML 檔案可在任何瀏覽器上檢視 (例如：Firefox、 Internet Explorer) ，
以及在一些文字處理系統上編輯(例如: OpenOffice Writer)。 
 

1.3.3.  改變GeoGebra的設定 

GeoGebra 在選項選單改變並儲存的偏好設定。舉例而言，可以改變’角的單位’從’角
度’到’弧度’，或著改變’點的樣式’、’勾選框大小’與’直角的樣式’。此外，可以改變坐標

的顯示方式與物件標籤的顯示。 
請參考“選項選單”章節以得到更多資訊。 
 
可以從選項選單選擇  “儲存設定”儲存的偏好設定。當執行後，GeoGbra 會記憶

的偏好設定並且在每個新開的 GeoGbra 檔案中使用它。 
註：可以由選項選單選擇“回復設定值”以回復原始設定。 
 
註： 如果利用 GeoGbra 作為展示工具，可以設定字體大小(選單選項)讓的觀眾可以方

便閱讀內文與物件標籤。 

1.4.  GeoGebra 作為編輯工具 

1.4.1.  列印選項 

列印幾何視區 

GeoGebra 列印在建構於幾何視區的圖形。可以在檔案選單中找到’預覽列印’。在預覽

列印出現的對話框中，可在此指定’名稱’、’作者’、’日期’。此外，可以設定列印’比例’
大小(以公分表示)以及改變列印的方向(直印或橫印)。  
 
註:：作任何改變之後按下 Enter 鍵以更新預覽視窗。 

列印構圖案本  

如果想要列印構圖案本，首先必須開啟構圖案本 (檢視選單)。然後，開啟構圖案本的

列印預覽視窗，可在新視窗的 檔案 選單中找到 列印預覽 選項。 
同樣的，可在此指定名稱、作者、日期。此外，在列印構圖案本之前可以設定列印’比
例’大小(以公分表示)以及改變列印的方向(直印或橫印)。 
 
註: 可以使用構圖案本視窗的檢視選單來切換 開/關 構圖案本中不同的欄位 名稱、定

義、指令、代數、暫停點。 
 

1.4.2.  建立幾何視區圖檔 

儲存幾何視區為圖檔  

可以將所建構的幾何視區以圖檔儲存於的電腦中。 
註：幾何視區將儲存為一個圖檔。如果的構圖並未使用所有的幾何視區域，將可以…  

 …使用工具 移動幾何視區、 放大、 縮小， 將圖形放置在幾何視區的左

上角位置，然後使用滑鼠拖曳GeoGebra視窗的角落來縮放大小。 
 …使用“選取區塊”以明確選出在幾何視區要被輸出或儲存成圖檔的區域。  
 可以建立兩個點稱為「輸出點1」與「輸出點2」，並將其定義為選取區塊的對

角線上兩點。  
註：「輸出點1」與「輸出點2」在幾何視區中必須在可看見的區域。  

 
在檔案選單中，選擇’輸出’後點擊’匯出圖檔'，在出現的視窗中可以指定輸出圖檔的’格
式’、’比例‘(以公分表示)及’解析度’(dpi)。  
 
註：輸出圖片的真實大小顯示於輸出視窗下方，同時以公分與像素單位呈現。 
 
更多關於圖檔的資訊請參考輸出幾何視區為圖檔小節。 

複製幾何視區到剪貼簿 

有很多種不同的方法可以複製幾何視區到電腦的剪貼簿：  
 在編輯選單，點擊  “複製到剪貼簿”。  
 在檔案選單，先選擇輸出，接著點擊  “複製到剪貼簿”。 
 在“輸出幾何視區為圖檔”的對話視窗(檔案選單─輸出─輸出幾何視區為圖檔

(png, eps)…)可以點擊“剪貼簿”按鈕。 
 
幾何視區的截圖可以複製到系統的剪貼簿成為 PNG 圖形(參考 PNG 格式)。此圖檔可

任意貼到其他程式中(例如: Microsoft Word 文件)。 
 
註: 為指定構圖的指定大小(以 cm)，請使用檔案選單中輸出選項的 ’匯出圖檔’(參見 
匯出圖檔)。 

1.4.3.  建立互動式網頁    

GeoGebra 可以建立互動式網頁，稱為動態工作單。在檔案選單中，點擊’輸出’，然後

點擊“匯出網頁(html)”，開啟動態工作單的輸出對話視窗： 
 在輸出的視窗的上方可以輸入動態工作單的’標題’、’作者’、’日期’。 
 切換’一般’，可以在動態構圖的前、後加入一些文字（例如，構圖和一些工作的

描述），構圖案本的本身也可直接置於網頁或以按鈕來開啟。 
 切換’進階’，可以改變動態構圖的功能(例如，重設圖示、按兩下滑鼠以開啟

GeoGebra應用視窗)，以及修改使用者介面(例如:顯示工具列、修改高度及寬

度)。 
註：如果應用程式(applet)的顯示範圍太大時，將會自動改變大小。 

 
註：輸出動態工作單時會產生三種檔案：  

 html 檔案, (例如：circle.html) - 此檔案會包括工作單本身。 
 ggb 檔案, (例如： circle.ggb) - 此檔案會包括GeoGebra構圖。 
 geogebra.jar (許多檔案)- 此檔案會包括GeoGebra以確保工作單可互動。  

這些檔案 (例如： circle.html、circlet.ggb 及 geogebra.jar files) 必須放在同一個資料

夾 (目錄) ，動態構圖才能順利執行。 
 
輸出的 HTML 檔 - （例如：circle.html） -可用任何瀏覽器觀看(例如：Mozilla, 
Internet Explorer)。為使動態構圖有效運作，電腦必須安裝 Java。免費的 Java 可以

從網站 http://www.java.com 下載。若想在學校網路的電腦上使用動態工作單, 請要求

學校的電腦網路管理員在電腦上安裝 Java。 
 
註：動態工作單的編輯，可以使用文字編輯器(例如：FrontPage、OpenOffice Writer)
來開啟輸出的 HTML 檔案。  

2.  幾何輸入 

2.1.  一般須知  

幾何視區 呈現數學物件的圖形表徵 (例，點、向量、線段、多邊形、函數、曲線、直

線及圓錐曲線)。 當滑鼠移到一物件上該物件會反白，且會出現該物件的說明。  
 
這裡有許多工具/模組告訴GeoGebra如何在幾何視區 (參考 錯誤! 找不到參照來源。)
輸入物件。例如：按下幾何視區可新增一點，(參見  新點工具 )，兩物件的交點(參見

交點工具) 或新增一個圓(參見   圓工具)。  
 

2.2.  構圖工具 

下列的構圖工具/模組可由點擊工具列的按鈕來啟動。另外，按下工具按鈕右下角的小

箭頭，可以開啟工具盒，有類似的工具。 
註：使用大部分的構圖工具，在幾何視區上空白處按滑鼠一下可以建立新的點。  

選取物件 

要選取物件，先點擊  移動工具，然後使用滑鼠選取物件。 
 
假如想要同時選取多個物件，可以拉出一個選取方塊：先點擊 移動工具，然後在要

選取方塊左上角按下滑鼠左鍵，然後持續按住滑鼠左鍵移動至欲選取方塊的右下角，

放開滑鼠鍵，所有在選取方塊內的物件都會被選取。 
 
註: 亦可藉由按住 Ctrl-鍵然後點擊不同的物件，則可以同時選取許多物件。 

物件快速更名 

欲變更選取物件或新增物件的名稱，只需開啟此物件重新命名 的對話方塊，然後輸入

被選取物件的新名稱並按下 ‘OK’ 按鈕。 

2.2.1.  一般工具 

 複製格式  

此工具可複製物件樣式 (例如：顏色、大小、線的樣式) 到一個或數個其他的物件。先

選取想要複製其樣式的物件，然後再點擊其它的物件。  

 刪除  

點擊欲刪除的物件。  

註: 若不小心刪錯物件，可以使用  ‘復原’ 按鈕。 

 移動  

使用滑鼠拖曳自變物件。在  移動模式，藉由滑鼠點擊一個物件，可以…  
 … 按下 Delete-鍵來刪除之 
 … 使用方向鍵來移動之 (參見 手動動畫)  

註: 按下 Esc-鍵，可以快速啟動  移動 工具。 

 移動幾何視區  

使用滑鼠拖曳幾何視區來改變視區範圍。 
註: 同時按 Shift-鍵(MS Windows: 也可用 Ctrl-鍵) 並以滑鼠拖曳來移動幾何視區。  
 
註: 在此模式下可使用滑鼠拖曳座標軸，來改變座標軸的比例。 

 記錄到試算表 

此工具能夠在物件移動的同時，並將該物件的變化數值記錄在試算表視區中。此工具

僅能使用在數值、點、向量。 
 
註: GeoGebra 會使用試算表的前兩個空白欄來記錄被選取物件的數值。 
 

 判斷物件關係  

選取兩物件在快顯示窗中可得其關係 (亦參見指令 Relation).  

 轉動  

先選取旋轉的中心點，然後滑鼠拖曳自變物件將會以此點為中心旋轉。 

 顯示或隱藏名稱  

點擊物件以顯示或隱藏其標籤。  

 顯示或隱藏物件  

在啟動此工具後，選取欲顯示或隱藏的物件。在切換到其它工具之後，物件的可見狀

態便會改變。 
 
註: 當啟動此工具時，在螢幕上被選取的所有物件將會被隱藏起來。 

 放大  

在幾何視區中任意處點擊滑鼠以拉近檢視視窗 (亦參見 自訂幾何視區)  

 縮小  

在幾何視區中任意處點擊滑鼠以縮小的構圖 (亦參見 自訂幾何視區)  

2.2.2.  點  

 交點  

兩物件的交點可由兩種方式產生，若… 
 … 選取兩物件，(盡可能)所有交點 會被建立。  
 … 直接點擊兩物件的交點，只建立此一交點。 

 
註: 對於線段、射線、或弧，可以指定是否要落在外部的交點 ，參考屬性對話方塊中

的一般。這可用於作出落在某物件的延伸處的交點，例如：線段或射線的延伸就是一

條直線。  

 中心點  

可以點擊兩點或一線段以得到中點。也可以點擊一圓錐曲線以建立該圓錐曲線的中心

點。 

 新點  

點擊幾何視區以建立新的點。  
註: 當滑鼠按鍵被放開時，此點的座標就會被固定。  
 
點擊一線段、直線、多邊形、圓錐曲線、函數或曲線，可在物件上新增一點 (亦參見指

令 點)。  
 
註: 點擊兩物件的相交處便建立此交點 (亦參見指令 交點).  

2.2.3.  向量  

 向量 (過兩點) 

選取向量的起點和終點。  

 向量 (指定起點、向量) 

選取點 A 及一向量 v 以建立點 B = A + v ，可以做出從 A 到 B 的向量。 

2.2.4.  線段 

 線段(過兩點)  

選取點 A 和點 B，以建立線段 AB。在代數視區會顯示線段的長度。 

 線段(指定起點與長度)  

點擊點 A 作為線段的起點 ，在出現的視窗中指定想要的長度。 
 

註: 此工具建立長度 a 且端點 B 的線段，並可用工具  移動 繞著起點 A 旋轉。.  

2.2.5.  射線  

 射線(過兩點)  

選取兩點 A 和 B，建立起點 A 通過 B 的射線， 在代數視區中顯示相關的直線方程

式。 

2.2.6.  多邊形  

 多邊形 

至少選取三個點作為多邊型的頂點。最後，點擊第一個點以形成一個封閉的多邊型，

在代數視區顯示多邊型的面積。 

 正多邊形 

選取兩點 A 和 B 並在出現的對談視窗中輸入一整數 n，即得到一個有 n 個頂點的正多

邊形(包括點 A 和 B)。 

2.2.7.  直線 

 角平分線  

有兩種方式可定義角平分線：  
 選取三點 A, B, C，並產生此三點圍成之角的角平分線，其中 B 點為頂點。  
 選取兩條線，並產生其兩角平分線。  

註： 所有的角平分線的方向向量長度都為 1。 

 最適直線 

以下的方法可建立一個點集的最適直線(亦參考指令 FitLine)： 
 建立 選取方塊 使其包含點集裡頭的所有點。 
 選取 點的串列 建立相對應於的最適直線。 

 直線(過兩點)  

選取兩點 A 和 B，建立直線通過點 A 和 B，此線的方向向量為(B - A)。  

 平行線 

選取一直線 g 和一點 A ，可畫出一直線通過 A 點且平行於 g，此線的方向向量為直線 
g 的方向向量。 

 中垂線 

點擊一線段 s 或兩點 A 和 B 以建立中垂(或垂直平分線)。 
註： 此線的方向向量等於線段 s 或 AB 的法向量(亦參見指令 PerpendicularVector)。 

 垂直線 

選取一直線 g 和一點 A，建立一直線通過 A 且垂直直線 g。 
 
註：此線的方向向量等於 g 之法向量(亦參見指令 PerpendicularVector).。 

 極線或徑線 

此工具建立一圓錐曲線的極線或徑線，可用… 
 …選取一點及圓錐曲線以獲得極線。  
 …選取直線或向量及一圓錐曲線以獲得徑線。  

 切線  

一圓錐曲線的切線可由兩種方式產生：  
 選取一點 A 及一圓錐曲線 c，並產生通過 A 且切於 c 的所有切線。 
 選取一線 g 及一圓錐曲線 c，並產生平行於 g 且切於 c 的所有切線。 

 
選取一個點 A 及一函數 f ，可產生 f 在 x = x(A) 的切線。  
註： x(A) 表視為 A 點的 x 座標。若 A 點在函數圖形上，則切線會通過 A 點。 

2.2.8.  圓錐曲線 

圓(指定圓心與半徑數值) 

選取圓心 M 後，在出現的視窗中輸入半徑。 

圓(指定圓心與一點)  

選取一點 M 和一點 P ，可畫出以 M 為圓心通過點 P 點的圓。  
 
註：此圓的半徑為 MP 的距離。  

圓(過三點)  

選取三點 A, B, C 可畫出通過此三點的圓。  
 
註：若這三點落在一直線上，此圓將退化成直線。 

圓(指定圓心，半徑長) 

英國：圓規  
 
選取一線段或兩點作為半徑，然後點擊一點作為新圓的圓心。 

圓錐曲線(過五點)  

選取五個點，產生通過此五點的圓錐曲線。  

註：若五個點中的其中四個點在同一直線時，圓錐曲線無定義。 

 橢圓 

選取兩點作為橢圓的焦點，然後選定第三點在此橢圓上。 

 雙曲線 

選取兩點作為雙曲線的焦點，然後選定第三點落在此雙曲線上。 

 拋物線  

選取一點和拋物線的準線。 

2.2.9.  圓弧與扇形 

註：圓弧的代數值即為其長度，扇形的代數值為其面積。  

圓弧(指定圓心與兩點)  

先選取圓弧的圓心 M 點，然後選取 起點為 A 終點為 B 的圓弧。  
 
註：點 A 落在圓弧上， 點 B 可以落在圓弧外。  

扇形(指定圓心與兩點)  

先選取扇形的圓心 M 點，然後選取起點為 A 終點為 B 的扇形。  
 
註：點 A 落在扇形的弧上， 點 B 可以落在圓弧外。 

圓弧(過三點)  

選取三點 A、B 和 C 建立通過三點的圓弧。點 A 為圓弧起點, 點 B 在圓弧上點 C 為圓

弧的終點。 

扇形(過三點)  

選取三點 A、B 和 C 建立扇形。點 A 為扇形圓弧起點, 點 B 在扇形圓弧上點 C 為扇形

圓弧的終點。 

半圓(過兩點)  

選取兩點 A 和 B，建立以線段 AB 為直徑的半圓。  

2.2.10.  數值與角度 

角度  

此工具可以建立 …  
 三點間的角度，第二個點為頂點 
 兩線段間的角度  
 兩直線間的角度 
 兩向量間的角度  
 多邊形的所有內角  

註: 若以反時針選取頂點所構成的多邊形，角度工具會給出多邊形內部的角度。 
 
註: 角度通常是逆時針產生。因此，選取物件的順序與角度是相關的。若要限制角度最

大為 180°，在 屬性對話框 的「一般」取消「角度可以大於 180°」的選項。 

指定角 

選取兩點 A 和 B ，並於出現的視窗的文字區塊輸入角度的大小，此工具建立點 C 及
角度 α，其中 α 為角 ABC。  

 面積 

此工具測量多邊形、圓、或橢圓的面積，並以動態文字顯示於幾何視區中。 

 距離或長度 

此工具可測量兩點、兩直線、或一點與一直線間的距離，並以動態文字顯示於幾何視

區中。亦可測量線段的長度、圓周長或多邊形周長。 

 數值滑桿  

註: 在 GeoGebra 中數值滑桿是自變數值或角度之圖形表徵。對於自變數值或角度可

簡單地建立數值滑桿(參考內容選單; 參考工具  顯示/隱藏 物件)。  
 
點擊幾何視區的任意位置可建立數值或角度的滑桿，在出現的視窗中，可以指定其名

稱、數值或角度的範圍區間[極小值, 極大值]，角度或長度的增量，滑桿水平或垂直，

以及寬度 (以點畫素為單位)。 

數值滑桿的位置在幾何視區可以是絕對位置(這表示數值滑桿不會因為放大縮小而受影

響，總是會在幾何視區上可看到的地方)或相對於座標系統(參考關於數值或角度的屬

性對話框)。  
 
註: 在數值滑桿的對話視窗上可以輸入 ° 或 π，另外在增量與範圍欄位上可以使用下列

快捷鍵： 
 Alt-O (MacOS: Ctrl-O) 可以出現角度符號 °  
 Alt-P (MacOS: Ctrl-P) 可以出現 pi 符號 π 

 斜率 

此工具可求出直線的斜率，並顯示於幾何視區中。 

2.2.11.  布林值 

顯示或隱藏顯示或隱藏群組物件之選項 

在幾何視區點擊滑鼠，會產生一個可以顯示或隱藏物件的核選方塊(參考 布林變數)。
在對話視窗中可以受此核選方塊影響的物件。 

2.2.12.  軌跡 

 軌跡 

選取一個會隨著點 A 而變動的點 B，然後點擊 A ，就會建立 B 的軌跡。  
註: 點 A 必須為一個物件上的一點 (例如： 直線、線段、圓)。 
 
範例:  

 在指令列上輸入 f(x) = x^2 – 2 x – 1 
 在 x 軸上取一新點 A (參考   新點; 參考指令點).  
 建立點 B = (x(A), f'(x(A)))，此點與點 A 相依。 
 選取工具  軌跡，然後點擊點 B 與點 A 
 沿著 x 軸拉動點 A，則會觀察到點 B 沿著軌跡移動。 

2.2.13.  幾何變換  

下列的幾何變換適用於點、直線、圓錐曲線、多邊形與圖片。  

 伸縮  

UK English: Enlarge Object from Point by Factor 

先選取進行伸縮的物件，再點擊伸縮中心，然後在出現的對話視窗的文字方塊輸入指

定伸縮倍率。  

 線對稱  

UK English: Reflect Object in Line 
 
先選取進行線對稱的物件，然後點擊直線指定為對稱線。  

 點對稱 Reflect Object about Point  

UK English: Reflect Object in Point 
 
先選取進行點對稱的物件，然後點擊點指定為對稱點。  

 反演 

UK English: Reflect Point in Circle 
 
此工具的功能可以讓一個點對一個圓進行反演。先選取進行反演的點，然後點擊圓指

定為反演圓。  

 旋轉  

先選取進行旋轉的物件，然後點擊點指定為旋轉中心，在出現的對話視窗的文字方塊

輸入旋轉角度。 

 平移  

先選取進行平移的物件，然後點擊平移的向量。  

2.2.14.  文字  

 插入文字  

使用此工具在幾何視區中產生靜態文字、動態文字、或 LaTeX 數學式。 
 
首先，需要指定文字的位置，可以下列其中一中方式建立： 

 點擊在幾何視區中任意區域，即可建立新的文字。 
 點擊某點建立新的文字黏貼在此點上。 

然後，可在顯示的對話框中輸入文字。 
註: 可以在基本屬性屬性對話框指定文字的位置在螢幕上對於座標系統為絕對或相對。  
 
動態文字 包含物件的數值。為了建立動態文字，可以使用鍵盤輸入靜態的部分(例， 
Point A =)，然後，點擊想要顯示其數值的物件。 
  
註: GeoGebra 會自動地增加必要的語法結構於動態文字中：文字的靜態部分會加上雙

引號並使用+這個符號來連結文字的不同部分。 
 

輸入 說明 
This is a text 呈現” This is a text”(靜態) 
"Point A = " + A 動態的呈現”Point A= ( 3.05, 2.54 )”  
"a = " + a + "cm" 動態的呈現” a = 5.87 cm” 

 
註: 若物件的名稱 xx 已經存在，可以嘗試建立靜態文字來當作物件的名稱，但需要使

用雙引號將其括起來 "xx"。否則，GeoGebra 會自動視為動態文字，呈現物件 xx 的

數值。然而，若輸入任何文字不是已存在的物件名稱，則不需要使用雙引號。 
 
註: 在動態文字中，靜態文字的部分必須使用雙引號括起來。文字不同的部分(例, 靜態

部分與動態部分)需要使用 +這個符號來連結。 

LaTeX 數學式  

在 GeoGebra 亦可書寫數學式。在  輸入文字 的對話視窗中勾選「LaTeX 數學

式」，並以 LaTeX 語法輸入數學式。  
 
註: 勾選 LaTeX 選項後，在下拉選單中可以選取一般的數學符號產生語法。然後，需

要依據公式語法在適當位置輸入相對應物件的名字(通常在一組大括號之間 { }，參考下

表） 
 
在下面表格說明一些重要的 LaTeX 指令，若需要更進一步的訊息請參考 LaTeX 說明

文件。 
 

LaTeX 輸入 輸出 
a \cdot b ba 

\frac{a}{b} 
b

a
 

\sqrt{x} x  

\sqrt[n]{x} n x  

\vec{v}  

\overline{AB} AB
x^{2} 2x  

a_{1} 1a  

\sin\alpha + 
\cos\beta 

 cossin   

LaTeX 輸入 輸出 

\int_{a}^{b} x dx 
b

a
xdx  

\sum_{i=1}^{n} i^2 

n

i
i

1

2
 

2.2.15.  圖片 

 插入圖片  

此工具可於幾何視區插入圖片 
首先，可以使用以下兩種方法其中一種來指定圖片的位置： 

 點擊在幾何視區，指定此點為圖片的左下角  
 點擊某點，以指定此點為圖片的左下角。 

然後出現檔案開啟的對話框，可在此選擇所要插入的圖檔。  
 
註: 在選擇此工具 後，可使用快捷鍵 Alt-click 將圖片從電腦的剪貼簿上直接貼至幾

何視區   。 

圖片的屬性 

位置 
圖片的位置可能是螢幕的絕對位置或相對於座標系統，可以在圖片屬性對談框之一般

進行設定。 
 
也可在屬性對談框之位置中指定三個頂點來設定，這可提供更彈性地縮放、旋轉，甚

至扭曲圖片。  
 

 角1：(圖片的左下角位置) 
 角2：(圖片的右下角位置) 

註: 此頂點只有當角1已事先設定時才能設定，可以用它來控制圖片的寬度。  
 角4(圖片的左上角位置) 

註: 此角只有當角1已事先設定時才能設定，可以用它來控制圖片的高度。  
註: 亦參見指令 Corner  
 
範例: 
新增三個點 A, B, 和 C  來探討頂點的影響。  

 設定點 A 為圖片的第一個頂點，點 B 為圖片的第二個頂點，在  移動模式，

拖曳點A及B ，將可以探索圖片會怎麼變化 
 設定點 A 為圖片的第一個頂點，點 C 為圖片的第四個頂點，並探索圖片會怎麼

變化  
 最後，設定這三個頂點，並觀察拖曳這些點會如何扭曲圖片  

 
範例: 

已經觀察到如何影響圖片的大小和位置。若想將圖片附著於點 A 並設定其寬度為 3，

高度為 4 單位時，可作下列設定：  
 設定’角1’為 A  
 設定’角2’為 A + (3, 0)  
 設定’角3’為 A + (0, 4)  

註: 現在若在  移動模式拖曳點 A，圖片大小不會改變。  

背景圖  

可以在屬性對話框中的’一般’指定某一圖片為背景圖。背景圖會放在座標軸的後面，且

無法用滑鼠來選取。  
 
註: 欲改變圖片的背景設定，從編輯選單中點選  屬性…來開啟屬性對話框。  

透明度 

為了讓在圖片後面的物件或座標軸可以看得到，圖片可以設定為透明。圖片的透明度

設定可以藉由屬性對話框的’樣式’來設定’填滿’的數值，從 0% 到 100%。  

3.  代數輸入 

3.1.  一般須知  

數學物件的代數表徵(例如，數值、座標、方程式)都會顯示在左邊的「代數視區」

中。若要建立或更改物件，可以透過 GeoGebra 視窗底部的「指令列」來直接輸入代

數式(參見 直接輸入; 參見 指令)。 
 
註:在指令列中輸入結束後，必須要按下 Enter 鍵。 

 
註: 在任何時候按下 Enter 鍵可以快速切換到指令列與幾何視區。不需要使用滑鼠先點

擊指令列即可於指令列中輸入指令。 
 

命名物件 

使用指令列列建立一個物件的時，可以給予物件特定的名稱： 
 點：在GeoGebra中， 點的命名是用大寫字母表示， 只要在坐標前面加上名稱

與等號即可。 
範例：C = (2, 4), P = (1; 180°), Complex = 2 + i  

 向量：為了區別點與向量，在Geogebra中使用小寫作為向量的命名。同樣的， 
在向量坐標的前面加上名稱與等號即可。 
範例：v = (1, 3), u = (3; 90°), complex = 1 – 2i 

 直線、圓、圓錐曲線：這些物件的命名是在方程式前面加上名稱與冒號即可。 
範例：g: y = x + 3, c: (x-1)^2 + (y – 2)^2 = 4,  
hyp: x^2 – y^2 = 2 

 函數：在這些函數的方程式前面命名即可，例如： f(x) = or g(x)=。 
範例： h(x) = 2 x + 4, q (x) = x^2, trig(x) = sin(x) 

 
註: 
˙ 若不手動輸入物件名稱，Geogebra 將直接依字母排序自動命名。 
 
˙ 若物件名稱包含下標可以使用底線來建立下標，例如: 輸入 A_1 或 S_{AB}可以得

到 A1 或 SAB 。 

更改數值  

有兩種方法可以改變自由物件的數值： 
 更改 物件的數值，在指令列輸入名稱與新的數值 (參見 直接輸入)  

範例：若想更改數值a = 3, 在指令列輸入 a = 5，並按下Enter 鍵。 

 編輯 代數表徵：使用工具  Move 並且在代數視區的物件點兩下， 即可打開

可編輯文字的區塊改變物件數值。按下Enter鍵輸入即可。 
 

註： 若直接更改自由物件的數值，則與應變物件的數值也會跟著相關的親屬物件更

改，或跟著其他相依物件重新定義。 

顯示指令列的輸入歷程紀錄 

先將游標移至指令列，可以使用鍵盤的上下鍵↑和↓，一步一步瀏覽先前輸入的指令。 
註： 點擊在指令列左邊的  標誌，會顯示指令列的說明。 

指令列插入 名稱、數值或物件的定義 

插入物件的名稱: 啟動工具  移動 並選取想要插入名稱的物件。然後按下鍵盤上的

F5 鍵。 
註: 在按下 F5 鍵之前，物件的名稱會被附加到在指令列所已輸入的方程式。 
 
插入物件的數值: 有兩種方式可以插入物件的數值(例, (1, 3), 3x – 5y = 12)。 

 在物件上點擊右鍵 (Mac OS: Ctrl-click) 並從出現的右鍵選單中選取項目  複製

到指令列。 
 啟動工具  移動 並選取想要插入數值的物件。然後，按下鍵盤上的 F4 鍵。 

註: 在按下 F4 鍵之前，物件的數值會被附加在指令列所已輸入的方程式。 
 
插入物件的定義: 有兩種方式可以插入物件的定義 (例, A = (4, 2), c = Circle[A, B]) 

 按著 Alt 點擊物件來插入物件的定義。 
 請動工具  移動 並選取想要插入定義的物件。然後按下鍵盤上的 F3 鍵。 

註: 在按下 F3 鍵之前，物件的定義會被附加在指令列所已輸入的方程式。 

3.2.  直接輸入 

GeoGebra 可以處理數值、角、點、向量、線段、直線、圓錐曲線、函數、和參數曲

線。可以在指令列輸入這些物件，輸入座標或方程式後按下 Enter-鍵。 

3.2.1.  數字和角 

數字 

可以利使用指令列建立一個數字，若只輸入一個數字，例如：3， GeoGebra 將會指

定一個小寫字母做為的名稱。若想要給數字指定一個明確的名稱，可以輸入名稱，接

著輸入等號及數字 (例：藉由輸入 r = 5.32 建立十進位數的 r). 
 
註： 在 GeoGebra 中，數字與角度使用點 ‘.’ 作為小數點。 
 
若要用到某些數學常數，如圓周率 π 或尤拉數 e，可藉由指令列右方的下拉式選單選

取或使用快捷鍵。 

註： 若變數 e 沒有被使用為現存物件的名稱。在新式子中出現的 e，會被 GeoGebra 
視為尤拉數。 

角 

角可輸入角度「°」 或徑度「rad」，對於徑度常數 π 是很有用的，也可以用 pi 輸入。 

 
註： 以使用以下這兩個快捷鍵輸入 ° 或 π ： 

 Alt-O (MacOS: Ctrl-O) 產生角度符號 ° 。 
 Alt-P (MacOS: Ctrl-P) 產生pi的符號 π 。 

 
範例: 角 α 可輸入角度(例,α= 60°) 或徑度 (α= pi/3)。 

 
註: GeoGebra 所有的內部運算都是使用徑度。角度符號「 °」只是代表「π/180」這

個常數的慣用符號而已。 

 
範例： 
˙ 若 a = 30 是一個數字，則 α = a° 將轉換數值 a 成為 α = 30°，不需要改變其數

值。 
˙若輸入 b = α / °，則將轉換角 α 轉換回數字 b = 30 不需要改變其數值。 

滑桿和方向鍵  

自變數字與角度在幾何視區中「滑桿」來呈現(參見工具滑桿)。在代數視區中，可使

用方向鍵改變「數字」與「角」的數值(參見手動動畫)。 

限制數值範圍  

自變數字與角度可以指定一個區間[min, max]，使用屬性對話框的滑桿來設定(亦參考

工具  滑桿)。 
註: 對於 應變角度 可以指定是否為可來回，可以從屬性對話框的一般設定。 

3.2.2.  點與向量  

點和向量可以用直角座標或極座標來輸入(參見數字與角)。 
註:英文「大寫」名稱代表「點」, 「小寫」名稱代表「向量」。 
 
範例： 

 在直角座標，輸入點P或向量v，使用P = (1, 0) or v = (0, 5)。  
 在極座標，輸入點P或向量，使用 P = (1; 0°) 或 v = (5; 90°)。 

註： 必須使用分號來區分極坐標的長度與角度。若沒有輸入角度的符號，

GeoGebra 將視此角為徑度。 
 
在 GeoGebra，也可以使用點與向量進行計算。 
範例: 

 建立點 A、B 的中點 M，在指令列輸入 M = (A + B) / 2 。 
 計算向量 v 的長度，在指令列輸入 length = sqrt(v * v)。 
 

3.2.3.  直線與座標軸  

直線 

直線可以使用「方程式」或「參數式」在指令列中輸入。在這兩種情況下, 被使用的變

數都必須是已定義的(例如：數字、點、向量)。 
 
註:直線名稱後面須加上「冒號」(:)。 
範例: 

 輸入 g : 3x + 4y = 2 ，可以得到直線g 。  
 先定義參數 t (如：t = 3)，再輸入參數式形式g : X = (-5, 5) + t (4, -3)。 
 先定義參數 m = 2 和 b = -1，然後輸入方程式 g: y = m x + b。 

坐標軸 

在指令中可以使用 xAxis 和 yAxis 來當作座標軸的名稱。 
範例: 指令 Perpendicular[A, xAxis] 可構造通過點 A 並垂直於 x 軸的垂直線。 

3.2.4.  圓錐曲線  

圓錐曲線，可以用 x 和 y 的二元二次方程式來輸入。已定義的變數(例如：數字、點、

向量) 可以被使用在方程式中。 
註：圓錐曲線的名稱在後面輸入冒號與方程式即可。 
 
範例： 

 橢圓： ell: 9 x^2 + 16 y^2 = 144 
 雙曲線： hyp: 9 x^2 – 16 y^2 = 144 
 拋物線： par: y^2 = 4 x 
 圓 c1： c1: x^2 + y^2 = 25 
 圓 c2： c2: (x–5)^2 + (y+2)^2 = 25 

 
註：若事先定義兩參數 a = 4 且 b = 3 ， 可輸入橢圓方程式為 ell: b^2 x^2 + a^2 y^2 = 
a^2 b^2。 

3.2.5.  x的函數 

可以用已定義的變數(例如：數字、點、向量)或函數來輸入一個新的函數。 
 
範例： 

 函數 f ：    f(x) = 3 x^3 – x^2 

 函數 g：   g(x) = tan(f(x)) 
 未命名之函數：   sin(3 x) + tan(x) 

 
註: 所有內定的函數 (例, sin, cos, tan)將會在內定函數與運算的章節中說明。 
 
在 GeoGebra 中可以使用指令求出函數的積分 和 微分 。 
註：可使用指令 f’(x) 或 f’’(x),…求出已定義的函數 f(x) 之一次微分與二次微分。 
範例：首先定義函數 f(x) = 3 x^3 – x^2，然後可輸入 g(x) = cos(f’(x + 2))得到函數 g。 
 
更多, 函數平移某個向量(參見指令 平移)，若函數是自變函數可以使用滑鼠移動。 

限制函數區間 

如果要限制函數的定義域為區間[a,b], 可以使用指令函數。 

3.2.6.  內定函數與運算  

建立數字、坐標或方程式(參見直接輸入) 時，可以利用以下內定函數與運算。 
註：輸入內定函數時需要加入括號，括號與函數之間不需要空白。 
 

運算 輸入 
加法 + 
減法 - 
乘法 * or 空白鍵 
內積 * or 空白鍵 
複數乘法 ⊗

除法 / 
次方 ^ or 2 
階層 ! 
Gamma 函數 gamma( ) 
括號 ( ) 
計算某點的 x-座標 x( ) 
計算某點的 y-座標 y( ) 
絕對值 abs( ) 
正負號 sgn( ) 
平方根 sqrt( ) 
立方根 cbrt( ) 
0 到 1 的隨機數 random( ) 

指數函數 exp( ) or ℯx 
自然對數 ln( ) or log( )
對數(以 2 為底) ld( ) 
常用對數(以 10 為底) lg( ) 
餘弦 cos( ) 

運算 輸入 
正弦 sin( ) 
正切 tan( ) 
反餘弦 acos( ) 
反正弦 asin( ) 
反正切 atan( ) 
雙曲餘弦 cosh( ) 
雙曲正弦 sinh( ) 
雙曲正切 tanh( ) 
反雙曲餘弦 acosh( ) 
反雙曲正弦 asinh( ) 
反雙曲正切 atanh( ) 
小於或等於之最大整數 floor( ) 
大於或等於之最小整數 ceil( ) 
四捨五入 round( ) 

3.2.7.  布林變數與運算 

在 GeoGebra 中，可使用布林變數 “true” 與 “false”。例如：在指令列中輸入 a = true 
或 b = false 並按下 Enter 鍵。 

核選方塊及方向鍵 

在幾何視區中，自變布林變數是以核選方塊的表徵來呈現(參見工具 顯示或隱藏群組

物件之選項)。 在代數視區中，若要改變布林變數，可以使用方向鍵 (參見手動動畫)。 
 
註: 也可使用數值(0 或 1)表示使用布林變數。這個功能搭配動態滑桿的動態速度可作

為動畫的開始與結束。在這個情形下，動畫按鈕只會顯示於幾何視區。 

布林運算 

在 GeoGebra 中可使用下列布林變數的運算與條件，而這些指令可以從指令列右方的

選取表列中選取，或使用鍵盤輸入： 

 
 選單 鍵入 範例 a、b 的類型 

等於 � 
== a ≟ b 或 a == 

b 

數值、點、直

線、圓錐曲線 

不等於 ≠ 
!= a ≠ b 或 a != 

b 
數值、點、直

線、圓錐曲線 
小於 < < a < b 數值 
大於 > > a > b 數值 

小於或等於 ≤ 
<= a ≤ b 或 a <= 

b 
數值 

大於或等於 ≥ >= a ≥ b 或 a >= 數值 

 選單 鍵入 範例 a、b 的類型 
b 

且 ∧ && a ∧ b 布林變數 

或 ∨ || a ∨ b 布林變數 

否 ¬ ! ¬a 或 !a 布林變數 

平行於 ∥  a ∥ b 直線 

垂直於 ⊥  a ⊥ b 直線 

3.2.8.  物件串列與運算 

若要建立一些物件(例如： 點、線段、圓)的串列，可以使用大括號。 
 
範例：  

 L = {A, B, C} 為包含三點A, B, C 的串列。 
 L = {(0, 0), (1, 1), (2, 2)} 為包含三個未命名點的串列。 

比較物件串列 

比較兩物件串列，可以使用下列語法結構： 
 list1 == list2：檢查兩串列是否相等並給予回傳 true 或 false。 
 list1 != list2：檢查兩串列是否不相等並給予回傳 true 或 false。 

串列的運算與函數 

註：若在串列上執行運算與函數運算，則將會得到一個新的串列。  
 
加法與減法的範例： 

 List1 + List2：將兩串列中相對應的元素相加。  
註： 此兩串列的長度必須相同。 

 List + Number：於串列內每個元素上加上此數字。 
 List1 – List2：將第一個串列的元素減去第二個串列內相對應的元素。 

註： 此兩串列的長度必須相同。 
 List – Number：於串列內每個元素上減去此數字。 

 
乘法與除法的範例： 

 List1 * List2：將兩串列中相對應的元素相乘。  
註： 此兩串列的長度必須相同。若兩串列內元素為矩陣，進行矩陣乘法運算。 

 List * Number：於串列內每個元素上乘以此數字。 
 List1 / List2：將第一個串列的元素除以第二個串列內相對應的元素。

註： 此兩串列的長度必須相同。 
 List / Number：於串列內每個元素上除以此數字。 
 Number / List：將此數字除以串列內的每個元素。 

其他範例： 
 List^2：將此串列內的每個元素平方。 
 sin(List) ：將此串列內的每個元素取sin函數。 

3.2.9.  矩陣物件與矩陣運算 

GeoGebra 也能支援矩陣，表示方法為矩陣的每一列。 

範例：在 GeoGebra 中, {{1, 2, 3}, {4, 5, 6}, {7, 8, 9}} 表示成矩陣 。 

矩陣運算 

加法及減法的範例： 
 Matrix1 + Matrix2：兩個相同大小的矩陣相對應的位置相加。 
 Matrix1 – Matrix2：兩個相同大小的矩陣相對應的位置相減。 

 
乘法的範例： 

 Matrix * Number：將此給定數字乘於每個矩陣的元素上。 
 Matrix1 * Matrix2：利用矩陣的乘法運算求出矩陣。 

註：第一個矩陣的列數與第二個矩陣行數必須相等。 
範例：輸入{{1, 2}, {3, 4}, {5, 6}} * {{1, 2, 3}, {4, 5, 6}} 
得到{{9, 12, 15}, {19, 26, 33}, {29, 40, 51}}。 

 2x2 Matrix * Point(或向量): 矩陣乘以給定的點或著向量將得到一個點。

  
範例：輸入{{1, 2}, {3, 4}} * (3, 4) 得到 A = (11, 25). 

 3x3 Matrix * Point(或向量): 矩陣乘以給定的點或向量將得到一個點。 
範例：輸入{{1, 2, 3}, {4, 5, 6}, {0, 0, 1}} * (1, 2) 得到 A = 
(8, 20)。 
註：這個特別的例子對於仿射變換(affine transformations)的齊次座標

(homogenous coordinates): (x, y, 1) 為一個點，(x, y, 0)為一個向量。這個例子

等同:  
{{1, 2, 3}, {4, 5, 6}, {0, 0, 1}} * {1, 2, 1}. 

 
其他範例： (參見 矩陣指令): 

 Determinant[Matrix]：計算矩陣的行列式值。 
 Invert[Matrix]：給定矩陣的反矩陣。 
 Transpose[Matrix]：給定矩陣的轉置矩陣。 

3.2.10.  複數與複數運算 

GeoGebra 不直接支援複數，但可以使用點來模擬複數運算。 
範例：若在指令列輸入複數 3 + 4i ，則在幾何視區可以得到一個點(3, 4)，此點的座

標在代數視區 顯視為 3 + 4i 。  

註: 在代數視區中任何的點可以複數形式顯示。開啟點的屬性對話框 並選取複數形

式。 
 
如果變數 i 尚未被定義，它將被認定為有序數對 i = (0, 1) 或複數 0 + 1i。也就是說，

在指令列中輸入變數 i 作為複數 (例： q = 3 + 4i) 。 
 
註：能將任何的點或著向量視為複數並將其顯示在代數視區中：在點/向量上點擊右鍵

(MacOS: Ctrl-click)並由屬性視窗中選擇複數。 
 
加法及減法的範例： 

  (2, 1) + (1, -2) 相同於 (2 + 1i) + (1 – 2i) 並給複數(3, -1) 也能

顯示為3 – 1i. 
 (2, 1) - (1, -2) 相同於 (2 + 1i) + (1 – 2i) 並給複數(1, 3) 也能顯

示為1 – 3i. 
 
乘法及除法的範例： 

  (2, 1) * (1, -2) 相同於 (2 + 1i) * (1 – 2i) 並給複數(4, -3) 也能

顯示為4 – 3i. 
 (2, 1) / (1, -2) 相同於 (2 + 1i) / (1 – 2i) 並給複數(0, 1) 也能顯

示為0 + 1i. 
註: 一般的乘法(2, 1)*(1, -2)為兩向量的係數乘機。 
 
其他範例： 
GeoGebra 也能作實數與複數的運算。 

 3 + (4, 5) 相同於3 + (4 + 5i) 並給複數(7, 5) 或 7 + 5i. 
 3 - (4, 5) 相同於3 - (4 + 5i) 並給複數(-1, -5) 或-1 - 5i. 
 3 / (0, 1) 相同於3 / (0 + 1i) 並給複數(0, -3) 或0 -3i. 
 3 * (1, 2) 相同於3 * (1 + 2i) 並給複數(3, 6) 或3 -6i. 

3.3.  指令  

使用指令可以產生新物件或修改物件。 
 
註: 指令的結果可能依據輸入的名稱來命名。參考下面的例子，新增的點被命名為 S。 
 
範例: 求兩直線 g 和 h 的交點時，可輸入： 
S = Intersect[g, h] (參考 指令 交點).  
 
註: 物件名稱可使用下標:輸入 A_1 可得到 A1 或輸入 S_{AB} 可得到 SAB.。 

指令自動化補齊 

當 GeoGebra 在指令列中輸入指令，軟體會嘗試著自動地補齊指令。這個意味著，在

指令列只要輸入指令前兩個字，GeoGebra 會顯示字母最接近的指令。 
 若這個指令就是所要的，按下 Enter 鍵，即接受建議 然後游標會移動到括弧內

部。 
 若這個指令不是所要的，可以繼續輸入，GeoGebra 會再提供更接近所輸入字

母之指令。 

3.3.1.  一般指令  

構圖步驟數  

ConstructionStep[]: 回傳目前的步驟數 Construction Protocol。 
ConstructionStep[Object]: 回傳指定物件目前的步驟數 Construction Protocol  

刪除  

Delete[Object]: 刪除一物件及所有與其相依的物件 

關係  

Relation[Object a, Object b]: 顯示一訊息框讓我們得知 Object a 和 Object b 
之間的關係。  
註: 此指令可得知是否一點在一直線上（或一圓錐曲線上）、一直線是否切於一

圓錐曲線、或一直線是否與一圓錐曲線相交, 兩物件是否相等。 

3.3.2.  布林指令 

If  

If[Condition, Object]: 若條件式判斷為真時，可得到 Object。若條件式判斷為

假時，為 undefined Object。. 
If[Condition, Object a, Object b]: 假如條件式判斷為真時，可得到 Object 

a，若為假時，可以得到 Object b。 

IsDefined  

IsDefined[Object]: 依據物件式否有定義，回傳真假值( true or false)。 

IsInteger  

IsInteger[Number]: 依據數值是否為整數，回傳真假值( true or false)。 

3.3.3.  數字指令  

分點比 

AffineRatio[Point A, Point B, Point C]: 回傳共線三點 A, B, C 的比例值

λ，這裡 C = A + λ * AB。 

面積  

Area[Point A, Point B, Point C, ...]: 計算點 A, B, C, …所圍成的多邊形

面積 。 
Area[Conic c]: 圓錐曲線 c 的面積(圓或橢圓) 
 
註: 若要計算兩個函數圖形之間的面積，需要使用積分指令。 

座標軸單位 

AxisStepX[]: 回傳目前 X 軸單位的數字大小。 
AxisStepY[]: 回傳目前 Y 軸單位的數字大小。 
 
註: 整合指令 Corner 與 Sequence ， 座標單位(AxisStep) 指令可以建立慣用座標。

(亦參考 Customizing Coordinate Axes and Grid). 

二項式係數 

BinomialCoefficient[Number n, Number r]: 計算‘n 選 r’的二項式係數。 

圓周長 

Circumference[Conic]: 回傳圓錐曲線的周長。  
註: 只有在圓或橢圓有定義。 

交比 

CrossRatio[Point A, Point B, Point C, Point D]: 計算出共線四個點 A, 
B, C, 與 D 的交比為 λ。  
λ = AffineRatio[B, C, D] / AffineRatio[A, C, D] 

曲率 

Curvature[Point, Function]: 計算出函數上指定點的曲率。 
Curvature[Point, Curve]: 計算曲線上指定點的曲率。 

距離  

Distance[Point A, Point B]: 兩點 A 與 B 之間的距離。 
Distance[Point, Line]:點和線之間的距離。  

Distance[Line g, Line h]: 兩直線 g 與 h 的距離。 
註: 兩相交直線，其直線之間的距離為 0。 

註: 亦參考工具  Distance or Length 

主軸  

FirstAxisLength[Conic]: 回傳圓錐曲線主軸的長度(譬如：橢圓長軸的一半) 

最大公因數  

UK English: HCF 
 
GCD[Number a, Number b]: 計算數字 a 與 b 的最大公因數。(UK-English: HCF = 

highest common factor) 
GCD[List of numbers]: 計算數列的最大公因數。(UK-English: HCF = highest 

common factor) 

整除法 

Div[Number a, Number b]: 計算數字 a 除 b 的商。 

積分 

Integral[Function, Number a, Number b]: 回傳在區間[a , b]函數的的積分

l 。  
註: 此指令會同時畫出函數與 X 軸間的區域.  

Integral[Function f, Function g, Number a, Number b]: 計算區間[a, 
b]，f(x)－g(x)的積分。   
註: 此指令會同時畫出函數 f 與 g 間的區域.  

 
註: 亦參考指令 Indefinite Integral 

疊代 

Iteration[Function, Number x0, Number n]: 給定起始值 x0 ，重複代入 f 
函數 n 次 。 
範例: 定義 f(x) = x^2 ，然後輸入指令 Iteration[f, 3, 2]則會得到結

果 (32)2 = 81。 

最小公倍數 

LCM[Number a, Number b]: 計算兩個數字 a 與 b 的最小公倍數。(UK English: 
LCM = lowest common multiple) 

LCM[List of numbers]: 計算數列的最小公倍數。(UK English: LCM = lowest 
common multiple) 

長度  

Length[Vector]: 向量的長度。 
Length[Point A]: 原點到點 A 的長度。  
Length[Function, Number x1, Number x2]: 函數 f 圖形，從 x1 到 x2 之間的

長度。 
Length[Function, Point A, Point B]: 函數 f 圖形上 A 點到 B 點之間的長

度。  
註: 假如給定的點不在函數圖形上，則點的 x 座標將會當作區間的值。 

Length[Curve, Number t1, Number t2]: 曲線在參數 t1 與 t2 之間的長度。 
Length[Curve c, Point A, Point B]: 曲線 c 在點 A 與 B 之間的曲線長度。 
Length[List]: L 串列的長度(串列的元素個數)。 
 

註: 亦參考工具  Distance or Length 
 

離心率 

LinearEccentricity[Conic]: 計算圓錐曲線的離心率。 
註: 離心率是圓錐曲線的中心與焦點的距離。 

下和  

LowerSum[Function, Number a, Number b, Number n]: 函數 f 在[a,b] 區間

以 n 個長條區求出的面積下和。 
註: 此指令會同時繪出這些下和的長方形。 

最大最小值 

Min[Number a, Number b]: 指 a 和 b 中最小者 
Max[Number a, Number b]: 指 a 和 b 中最大者 

模數 

Mod[Integer a, Integer b]: 整數 a 除以整數 b 的餘數 

參數 

Parameter[Parabola]: 回傳拋物線 p 的參數(準線和焦點間的距離) 

周長 

Perimeter[Polygon]: 回傳多邊形的周長 

半徑  

Radius[Circle]: 回傳圓 c 的半徑 

亂數指令  

RandomBetween[Min integer, Max integer]: 在最大值與最小值區間，產生

的亂數。 
RandomBinomial[Number n of trials, Probability p]: 使用二項式分佈

(次數 n,機率 p)產生的亂數。 
RandomNormal[Mean, Standard deviation]: 使用常態分佈(平均值 m, 標準差

s)產生的亂數。 
RandomPoisson[Mean]: 使用卜瓦松分佈(平均值 m)產生的亂數。 

副軸長  

SecondAxisLength[Conic]: 回傳圓錐曲線副軸的長度(譬如：橢圓短軸的一半) 

斜率  

Slope[Line]: 回傳給定直線的斜率 
 
註: 此指令會同時畫出一個「斜率三角形」，參考 屬性對話方塊 的‘樣式’。 
註:亦參考工具  斜率 
 

梯形和  

UK English: TrapeziumSum 
 
TrapezoidalSum[Function, Number a, Number b, Number n of 

trapezoids]: 計算函數 f 在區間 [a, b] 分成 n 個梯形的總和。  
 
註: 此指令會同時畫出函數下方的梯形。 

上和  

UpperSum[Function, Number a, Number b, Number n]: 計算函數 f 在[a,b] 
區間以 n 個長條區求出的面積上和。  

 
註: 此指令亦可繪出這些上和的長方形。 

3.3.4.  角指令  

角  

Angle[Vector v1, Vector v2]:回傳兩向量 v1 和 v2 間的夾角(介於 0 and 360°)  
Angle[Line g, Line h]: 回傳兩直線 g 與 h 之方向向量間的夾角(介於 0 and 

360°)  
Angle[Point A, Point B, Point C]: 回傳 BA 和 BC 間的夾角(介於 0 到 360°), 

B 為頂點。  

Angle[Point A, Point B, Angle α]: 回傳以線段 AB 為始邊，畫出大小為 α 的

角。 
註: 與點 Rotate[A, α, B] 是相同的。 

Angle[Conic]: 回傳圓錐曲線 c 主軸的轉角(參見指令座標軸)。 
Angle[Vector]: 回傳 x-軸到向量 v 之間的夾角。 
Angle[Point]: 回傳 x-軸到點 A 的夾角。 
Angle[Number]: 將一數字 n 轉換成角(結果介於 0 到 2pi 之間)  
Angle[Polygon]: 建立多邊形 p 的所有內角(逆時針).  

註: 若多邊形是以逆時針方向選取，可以得到內角和。假如多邊形是順時針方向

選取，則可以得到外角和。 
註: 亦參考工具  角 與  角(給定角度)  
 

3.3.5.  點指令  

中心點 

UK English: Centre 
 
Center[Conic]: 回傳圓錐曲線的中心點。   

註: 只有在圓、橢圓與雙曲線有意義。 
註: 亦參考工具  中點 

重心  

Centroid[Polygon]: 回傳多邊形 p 的重心 

Corner 

Corner[Number n of Corner]: 產生幾何視區角落的座標值 (n = 1, 2, 3, 4)。
Corner[Image, Number n of corner]: 產生圖片角落的座標值 (n = 1, 2, 
3, 4)  

Corner[Text, Number n of corner]: 產生文字方塊角落的座標值(n = 1, 2, 3, 
4)  
 

註: n=1 表左下角，n=2 表右下角，n =3 表右上角，n=4 表左上角 

極值  

UK English: TurningPoint 
 
Extremum[Polynomial]: 計算多項式函數圖形上的局部極值。 

焦點  

Focus[Conic]: 產生圓錐曲線 c (全部的)焦點 

反曲點  

InflectionPoint[Polynomial]: 多項式 f 的所有反曲點.。 

交點  

Intersect[Line g, Line h]: 直線 g 和 h 的交點 
Intersect[Line, Conic]: 直線 g 和圓錐曲線 c 的所有交點(最多 2 個) 
Intersect[Line, Conic, Number n]: 直線 g 和圓錐曲線 c 的第 n 個交點。 
Intersect[Conic c1, Conic c2]: 兩圓錐曲線 c 和 d 的所有交點(最多 4 個)  
Intersect[Conic c1, Conic c2, Number n]: 兩圓錐曲線 c 和 d 的第 n 個 
交點  
Intersect[Polynomial f1, Polynomial f2]: 多項式 f1 和多項式 f2 的所有

交點  
Intersect[Polynomial f1, Polynomial f2, Number n]: 多項式 f1 和多項

式 f2 的第 n 個交點 
Intersect[Polynomial, Line]: 多項式 f 和直線 g 的所有交點  
Intersect[Polynomial, Line, Number n]: 多項式 f 和直線 g 的第 n 個交點 
Intersect[Function f, Function g, Point A]: 函數 f 和 g 在起始點 A 的

所有交點(牛頓法)  
Intersect[Function, Line, Point A]: 函數 f 和直線 g 在起始點 A 的所有交

點(牛頓法)  
 
註: 參考工具  交點 

中點  

Midpoint[Point A, Point B]: A 和 B 的中點  
Midpoint[Segment]: 線段 s 的中點 
註: 亦參考工具  中點 
 

點  

Point[Line]: 回傳直線上一點。 
Point[Conic]: 回傳圓錐曲線上一點(例如: 圓、橢圓、雙曲線)。 
Point[Function]: 回傳 f 函數上一點。 
Point[Polygon]: 回傳多邊形 p 上一點。 
Point[Vector ]: 回傳向量 v 上一點。 
Point[Point, Vector]: 回傳從點 P 平移向量 v 之後的點。 
註: 亦參考工具  新點 

根  

Root[Polynomial]: 多項式 f 的所有根 
Root[Function, Number a]: 函數 f 在起始值 a 的一個根(牛頓法) 

Root[Function, Number a, Number b]: 函數 f 在 [a, b]區間的根 (使用「錯位

法」：FalsePosition Method)  

頂點  

Vertex[Conic]: 產生圓錐曲線(全部的)頂點。 

3.3.6.  向量指令 

曲率向量 

CurvatureVector[Point, Function]: 函數 f 在點 A 的曲率向量。 
CurvatureVector[Point , Curve]: 曲線 c 在點 A 的曲率向量。 

方向向量 

Direction[Line]: 直線 g 的方向向量 
 
註: 直線方程式 ax + by = c 之方向向量為 (b, - a)。 

法向量 

PerpendicularVector[Line]: 直線 g 的法向量  
註: 直線方程式 ax + by = c 的法向量為(a, b)。 

PerpendicularVector[Vector v]: 向量 v 的法向量。   
註: 向量(a, b)的法向量為(- b, a)  

單位法向量 

UnitPerpendicularVector[Line]: 直線 g 的單位法向量。 
UnitPerpendicularVector[Vector]: 向量 v 的單位法向量。 

單位向量  

UnitVector[Line]: 直線 g 之單位方向向量 
UnitVector[Vector]: 與向量 v 同方向的單位向量 

向量  

Vector[Point A, Point B]: 從點 A 到點 B 的向量  
Vector[Point]: 點 A 的位置向量 
 

註: 亦參考工具  向量 

3.3.7.  線段指令  

線段 

Segment[Point A, Point B]: 建立點 A 和 B 間的線段 
Segment[Point A, Number a]: 建立以 A 為起點，線段長為 a 的線段  

註: 會同時產生另一個線段端點。 

3.3.8.  射線指令 

射線  

Ray[Point A, Point B]: 起點 A 通過 B 點的射線。 
Ray[Point, Vector v]: 起點 A 且方向向量為 v 的射線。 
註: 亦參考工具  射線(通過兩點) 
 

3.3.9.  多邊形指令 

多邊形 

Polygon[Point A, Point B, Point C,...]: 由給定點 A, B, C,…所圍成的多

邊形 
Polygon[Point A, Point B, Number n]: 有 n 個頂點的正多邊形(包括 點 A 和 

B) 
註: 亦參考工具  多邊形 與  正多邊形 
 

3.3.10.  直線指令 

角平分線  

AngleBisector[Point A, Point B, Point C]: 角 ABC 的角平分線 
註: B 為角的頂點。 

AngleBisector[Line g, Line h]: 直線 g 和 h 的角平分線。 
註: 亦參考工具  角平分線 
 

漸近線 

Asymptote[Hyperbola]: 雙曲線 h 的兩漸近線 

對稱軸 

Axes[Conic]: 圓錐曲線 c 的對稱軸 

徑線 

Diameter[Line, Conic]: 圓錐曲線 c 平行於直線 g 的徑線 
Diameter[Vector, Conic]: 圓錐曲線 c 平行於向量 v 的徑線 

準線 

Directrix[Parabola]: 拋物線 p 的準線 

主軸 

FirstAxis[Conic]: 圓錐曲線 c 的長軸或貫軸 

直線 

Line[Point A, Point B]: 建立 A 點和 B 點的直線。 
Line[Point, Line]: 建立 A 點且平行於 g 的直線。 
Line[Point, Vector v]: 建立點 A 且方向為 v 的直線。 
註: 亦參考工具  直線 

垂直線 

Perpendicular[Point, Line]: 建立 A 且垂直於 g 的直線。 
Perpendicular[Point, Vector]: 建立點 A 且垂直於向量 v 的直線。 
註: 亦參考工具  垂直線 
 

 中垂線 

PerpendicularBisector[Point A, Point B]: 建立段 AB 垂直的平分線。 
PerpendicularBisector[Segment]: 建立線段垂直的平分線。 
註: 亦參考工具  中垂線 
 

極線 

Polar[Point, Conic]: 建立相對於圓錐曲線 c 的極線。 
註: 亦參考工具  極線 
 

副軸 

SecondAxis[Conic]: 建立圓錐曲線 c 的短軸或共軛軸。 

切線 

Tangent[Point, Conic]: 建立圓錐曲線 c 過點 A 的(全部的)切線。 
Tangent[Line, Conic]: 建立圓錐曲線 c 的所有平行於直線 g 的切線。 
Tangent[Number a, Function]: 建立 f(x)在 x=a 時的切線。 

Tangent[Point A, Function]: 建立 f(x)在 x=x(A)時的切線。 
Tangent[Point, Curve]: 建立曲線 c 過點 A 的切線。 
註: 亦參考工具  切線 
 

3.3.11.  圓錐曲線指令 

圓 

Circle[Point M, Number r]: 圓心 M 且半徑為 r 的圓。 
Circle[Point M, Segment]: 圓心 M 且半徑為 s 的長度。 
Circle[Point M, Point A]: 圓心 M 且通過點 A 的圓。 
Circle[Point A, Point B, Point C]: 通過三點 A、B、C 的圓。 

註: 亦參考  Compass,  Circle with Center through Point,  Circle with Center 
and Radius, and  Circle through Three Points 

圓錐 

Conic[Point A, Point B, Point C, Point D, Point E]: 回傳通過點 A、

B、C、D、E 五點的圓錐曲線。 
註: 如果有四點成一直線，則無法做出此圓錐曲線。 

註: 亦參考工具  圓錐曲線(通過五點) 
 

橢圓 

Ellipse[Point F, Point G, Number a]: 建立焦點為 F,G 且長軸半長為 a 的

橢圓。 註: 2a 必須大於 FG 的距離。 
Ellipse[Point F, Point G, Segment]: 建立焦點為 F,G 且長軸半長等於線段

s 的長度的橢圓。 
Ellipse[ Point A, Point B, Point C]: 建立焦點為 A,B 且通過點 C 的橢

圓。 
註: 亦參考  橢圓 
 

雙曲線 

Hyperbola[Point F, Point G, Number a]: 建立焦點為 F,G 且長軸半長為 a
的雙曲線 。 
註: 條件: 0 < 2a < FG 的距離。 

Hyperbola[Point F, Point G, Segment]: 建立焦點為 F,G 且長軸半長等於線

段 s 的長度的雙曲。 
Hyperbola[Point A, Point B, Point C]: 建立焦點為 A,B 且通過點 C 的雙曲

線。 
註: 亦參考  雙曲線 

密切圓 

OsculatingCircle[Point, Function]: 函數 f 在點 A 的密切圓。 
OsculatingCircle[Point, Curve]: 曲線 c 在點 A 的密切圓。 

拋物線 

Parabola[Point F, Line g]: 回傳焦點為 F 且準線為 g 的拋物線。 
註: 亦參考  拋物線 
 

3.3.12.  函數指令 

條件函數 

若要建立條件函數可以使用布林指令 If(參見指令 If) 來建立函數。 
註: 可以對這些條件函數進行微分 derivatives 和積分 integrals，並求出與其他函數的

交點，和一般的函數一樣。 
 
範例:  

f(x) = If[x < 3, sin(x), x^2] 此函數相當於 
  sin(x) 當 x < 3  
  x2 當 x ≥ 3. 
 a ≟ 3 ˄ b ≥ 0 檢驗 a 是否等於 3 和 b 是否大於等於 0  

 
註: 條件敘述的符號(如 ≟, ˄, ≥)可以在指令列下拉列表中找到。 

導函數(微分) 

Derivative[Function]: 回傳函數 f(x)的微分 
Derivative[Function, Number n]: 回傳函數 f(x)的 n 次微分 
 
註: 可以用 f'(x)代替  Derivative[f],也可用 f''(x)代替 Derivative[f, 2]  

展開式 

Expand[Function]: 將式子的括號乘開。 
範例: Expand[(x + 3)(x - 4)] 會得到 f(x) = x2 - x - 12 

因式 

UK English: Factorise 
 
Factor[Polynomial]: 多項式的因式 。 

範例: Factor[x^2 + x - 6] 會得到 f(x) = (x-2)(x+3) 

函數 

Function[Function, Number a, Number b]: 產生函數圖形，將函數 f 的定義

域限制於 [a, b] 區間 。 
註: 這個指令如能使用在顯示在某個區間的函數 
範例: f(x) = Function[x^2, -1, 1]得到函數 x2 的圖形，定義域限制在[-1, 1]
區間。如果輸入 g(x) = 2 f(x),將可以得到函數 g(x) = 2 x2,但這個函數的定義域並

不限制於 [-1, 1]區間。 

積分 

Integral[Function]: 函數 f(x)的不定積分。 
 
註: 亦參考指令 Definite integral  

多項式 

Polynomial[Function]: 函數 f 的展開式  
範例: Polynomial[(x - 3)^2] 產生 x2 - 6x + 9  

Polynomial[List of n points]: 建立經過 n 個點的 n-1 次內插多項式。 

簡化 

Simplify[Function]: 簡化給定函數的項。 
範例:   

Simplify[x + x + x] 產生 f(x) = 3x  
Simplify[sin(x) / cos(x)] 產生 f(x) = tan(x)  
Simplify[-2 sin(x) cos(x)] 產生 f(x) = sin(-2 x) 

泰勒展開式 

TaylorPolynomial[Function, Number a, Number n]: 建立函數 f(x)對點

x=a 的 n 次泰勒展開式  

3.3.13.  參數曲線指令 

曲線  

Curve[Expression e1, Expression e2, Parameter t, Number a, 
Number b]: 產生座標為(e1, e2)的曲線，其中 e1，e2 為參數式 (使用參數 t
時，t 必需在 [a, b]區間 
範例: c = Curve[2 cos(t), 2 sin(t), t, 0, 2 pi] 

 
註:參數曲線可以像函數一樣使用  
範例: 輸入 c(3)會回傳 t=3 時的點座標   

註: 使用  新點工具在曲線上放一個新點(參見模組  新點，亦參見指令 點).由於參

數 a 和 b 為動態的，可在此處使用滑桿變數(參見工具  滑桿). 

參數曲線相關指令  

Curvature[Point, Curve]: 計算曲線上一點的曲率。 
CurvatureVector[Point, Curve]: 產生曲線上一點的曲率向量。 
Derivative[Function]: 回傳函數 f(x)的微分 
Derivative[Function, Number n]: 回傳函數 f(x)的 n 次微分 
Length[Curve, Number t1, Number t2]: 曲線在參數 t1 與 t2 之間的長度。 
Length[Curve c, Point A, Point B]: 曲線 c 在點 A 與 B 之間的曲線長度。 
OsculatingCircle[Point, Curve]: 曲線 c 在點 A 的密切圓。 
Tangent[Point, Curve]: 建立曲線 c 過點 A 的切線。 

3.3.14.  圓弧與扇形指令 

註: 圓弧的代數值為其長度，扇形的代數值為其面積。  

弧 

Arc[Conic, Point A, Point B]: 回傳介於圓錐曲線 c 上的兩點 A,B 間的弧 
  
註: 這個指令只能用於圓或橢圓  

Arc[Conic, Number t1, Number t2]: 回傳介於圓錐曲線 c 上兩參數 t1 和 t2 值

間的弧，圓錐曲線 c 為下列參數式:  
 圓: (r cos(t), r sin(t)) 其中 r 為圓的半徑 
 橢圓: (a cos(t), b sin(t)) 其中 a 和 b 為長軸和短軸半長  

圓弧 

CircularArc[Point M, Point A, Point B]: 建立圓心為 M，起點為 A，終點

為 B 的圓弧    
註: 點 B 不需落在圓弧上。 

註: 亦參考  Circular Arc with Center between Two Points 

扇形 

CircularSector[Point M, Point A, Point B]: 圓心為 M，起點為 A，終點

為 B 的扇形 
註: 點 B 不需落在弧上。  

註: 亦參考  Circular Sector with Center between Two Points 

CircumcircularArc  

CircumcircularArc[Point A, Point B, Point C]: 通過三點 A,B,C 的圓弧 

CircumcircularSector  

CircumcircularSector[Point A, Point B, Point C]: 通過三點 A,B,C 的

扇形 

扇形  

Sector[Conic, Point A, Point B]:介於圓錐曲線 c 上的兩點 A,B 間的圓錐曲

線扇形區  
註: 這個指令只能用於圓或橢圓. 

Sector[Conic, Number t1, Number t2]: 介於圓錐曲線 c 上兩參數 t1 和 t2 值

間的圓錐曲線扇形區，圓錐曲線 c 為下列的參數式:  
 圓: (r cos(t), r sin(t)) 其中 r 為圓的半徑 
 橢圓: (a cos(t), b sin(t)) 其中 a 和 b 為長軸和短軸半長 

半圓  

Semicircle[Point A, Point B]: 建立線段 AB 上的半圓。 
註: 亦參考工具  半圓 
   

3.3.15.  文字 

分數文字  

FractionText[Number]: 將數值轉換成以分數形式呈現的文字，如同顯示在

Graphics View 的(LaTeX)文字物件 
範例: 如果 a: y = 1.5 x + 2 是一條直線, 那麼 FractionText[Slope[a]]會

得到分數 文字 3/2. 

LaTeX 

LaTeX[Object]: 將物件回傳成 LaTeX 文字形式 
範例: 如果 a = 2 且 f(x) = a x2, 那麼 LaTeX[f] 回傳 2 x2 (如同 LaTeX text) 

LaTeX[Object, Boolean]: 將物件回傳成 LaTeX 文字形式。若布林值為”真”，則

數值取代變數，否則呈現變數名稱。 
範例: 如果 a = 2 and f(x) = a x2, 那麼  
LaTeX[f, true] 回傳 2 x2 (as a LaTeX text)  
LaTeX[f, false] 回傳 a x2 (as a LaTeX text) 

字元轉換成萬國碼 

LetterToUnicode["Letter"]: 將字元轉換成萬國碼 
註: 字元需要在引號的之間 
範例: LetterToUnicode["a"] 回傳 97 



51 
 

名稱  

Name[Object]: 回傳在幾何視區的物件的名稱 
註: 在物件的動態文字中使用此指令可能會重新命名。命名指令與 Object 指令

相反  

物件  

Object[Name of object as text]:回傳物件的名稱。 Returns the object for a 
given name which is given as a text (static and/or dynamic).   
註:  物件指令 與 Name 指令相反.  
範例: 如果有點 A1, A2, ... , A20 且滑桿 n = 2, 那麼 Object["A" + n] 給 點 
A2. 

TableText  

TableText[List 1, List 2, List 3,...]: 產生串列物件的表格 
註: 電腦預設每一串列產生表格中的一列。  
範例:   
TableText[{x^2, x^3, x^4}] 產生串列的表格。每列是由左到右排列 
TableText[Sequence[i^2, i, 1, 10]] 產生序列的表格。每列是由左

到右排列 
TableText[List 1, List 2, List 3,..., "Alignment of text"]: 產生

一個 text 包含表格內所列的物件。選擇的 text 即 Alignment of text 控制表格

text 的方向和排列。 
註: 可能的值為 "vl", "vc", "vr", "v", "h", "hl", "hc", "hr". 預設值為 "hl". 

"v" = 垂直, i.e. lists are columns  
"h" = 水平, i.e. lists are rows  
"l" = 從左到右排列  
"r" = 從右到左排列  
"c" = 中間  

範例: 
TableText[{1,2,3,4},{1,4,9,16},"v"] 產生兩行四列的 text，其中每

個元素都是從左到右排列 
TableText[{1,2,3,4},{1,4,9,16},"h"] 產生四行兩列的 text，其中每

個元素都是從從左到右排列 
TableText[{11.2,123.1,32423.9,"234.0"},"r"] 產生一列的 text，
其中每個元素都是從右到左排列 

文字  

Text[Object]: 產生物件的方程式如同文字物件 
註: 預設值, 數值都被變數所替代  
範例: 如果 a = 2 和 c = a2, 那麼 Text[c] 回傳文字 "4". 

Text[Object, Boolean]: 產生物件的方程式如同文字物件。如果數值取代變數，

那麼布林變數回傳”真” ；如果仍是變數，則回傳”假”  
範例: 如果 a = 2 和 c = a2, 那麼   

Text[c, true] 產生文字 "4".  
Text[c, false] 產生文字 "a2" 

Text[Object, Point]: 產生物件的公式如同文字物件在給定點的位置上  
範例: Text["hello", (2, 3)] “hello”畫在座標(2,3)的位置 

Text[Object, Point, Boolean]: 產生物件的公式如同文字物件在給定點的位置

上。如果數值取代變數，那麼布林變數回傳”真” ；如果仍是變數，則回傳”假” 

文字轉成萬國碼 

TextToUnicode["Text"]: 將 text 轉換成一串萬國碼  
範例:  
TextToUnicode["Some text"]會得到一串萬國碼  
{83, 111, 109, 101, 32, 116, 101, 120, 116}.  
如果 text1 是 "hello", 那麼 TextToUnicode[text1] 會得到   
{104, 101, 108, 108, 111}. 

萬國碼轉成字元 

UnicodeToLetter[Integer]: 將整數的萬國碼轉換回 Graphics View 中的字母 
範例: UnicodeToText[97] 會得到 text "a". 

萬國碼轉成文字 

UnicodeToText[List of Integers]: 將整數的萬國碼轉換回文字 
範例: UnicodeToText[{104, 101, 108, 108, 111}] 會得到 text "hello". 

3.3.16.  軌跡指令 

軌跡 

Locus[Point Q, Point P]: 點 Q 的軌跡線(P 為控制點)  
註: 點 P 必須為物件(如直線、線段、圓)上的一點 

3.3.17.  串列與序列指令 

Append 附加 

Append[List, Object]: 將物件加入串列中成為新串列的最後一個元素  
例: Append[{1, 2, 3}, (5, 5)] 可得到 {1, 2, 3, (5, 5)} 

Append[Object, List]: 將物件加入串列中成為新串列的第一個元素  
例: Append[(5, 5), {1, 2, 3}] 可得到 {(5, 5), 1, 2, 3} 

CountIf 符合條件的數目 

CountIf[Condition, List]: 計算串列中滿足條件的元素個數 
例: 

CountIf[x < 3, {1, 2, 3, 4, 5}] 可得到數值 2  
CountIf[x<3, A1:A10] 計算在試算表中 A1:A10 格中數值小於 3 的格數。 

元素 

Element[List, Number n]: 呼叫串列中的第 n 個元素  
註: 串列僅能有一種物件形式。(例, 僅能數字或點). 

First 

First[List]: 回傳串列中的第一個元素 
First[List, Number n of elements]: 回傳一個包含給定串列的前 n 個元素的

新串列。 

插入 

Insert[Object, List, Position]: 插入物件到串列中指定的位置。  
例: Insert[x^2, {1, 2, 3, 4, 5}, 3] 將 x2 插入到串列中的第三個位

置而得到新串列 {1, 2, x2, 3, 4, 5}  
註: 若指定的位置為負值，則位置由右邊算起。  
例: Insert[(1, 2), {1, 2, 3, 4, 5}, -1] 將點插入到串列的最後一

個位置而得到新串列 {1, 2, 3, 4, 5, (1, 2)} 
Insert[List 1, List 2, Position]: 將 list1 的所有元素插入到 list2 中指定的

位置。  
例: Insert[{11, 12}, {1, 2, 3, 4, 5}, 3] 將 list1 所有元素插入到 
list2 的第三個 (與隨後的) 位置並得到新串列 {1, 2, 11, 12, 3, 4, 5}  
註: 若指定的位置為負值，則位置由右邊算起。  
例: Insert[{11, 12}, {1, 2, 3, 4, 5}, -2] 將 list1 所有元素插入到

list2 的倒數第二個位置並得到串列 
{1, 2, 3, 4, 11, 12, 5}. 

交集 

Intersection[List 1, List 2]: 可得到一個新串列包含兩串列中都有的元素 

迭代序列 

IterationList[Function, Number x0, Number n]:   
可得到 n+1 個元素的串列，其中元素是由函數代入起始值 x0 經過多次迭代而

組成的  
例: 在定義函數 f(x) = x^2 之後，指令  
L = IterationList[f, 3, 2] 可得到串列   
L = {3, 32, (32)2} = {3, 9, 81} 

合併 

Join[List 1, List 2, ...]: 合併兩個 (或多個) 串列   
註: 新串列包含所有元素，即使具有相同的元素。新串列的元素並不會重新排

列。  
例:  
Join[{5, 4, 3}, {1, 2, 3}] 可得串列 {5, 4, 3, 1, 2, 3} 

Join[List of lists]: 將子串列合併得到一個更大的串列  
註: 新串列包含所有元素，即使具有相同的元素。新串列的元素並不會重新排

列。  
例:  
Join[{{1, 2}}] 可得串列 {1, 2} 
Join[{{1, 2, 3}, {3, 4}, {8, 7}}] 可得串列  
{1, 2, 3, 3, 4, 8, 7} 

KeepIf 

KeepIf[Condition, List]: 產生一個新串列包含滿足條件的元素  
例: KeepIf[x<3, {1, 2, 3, 4, 1, 5, 6}] 可得到新串列 {1, 2, 1} 

Last 

Last[List]: 回傳串列中的最後一個元素。 
Last[List, Number n of elements]: 回傳一個包含給定串列的最後 n 個元素

的新串列。 

長度 

Length[List]: 得到指定串列的長度，也就是元素個數。 

取最小元素 

Min[List]: 回傳串列中最小的元素 

取最大元素 

Max[List]: 回傳串列中最大的元素 

內積 

Product[List of numbers]: 計算串列中所有數值的乘積 

RemoveUndefined 

RemoveUndefined[List]: 移除串列中未定義的物件。  
例: RemoveUndefined[Sequence[(-1)^i, i, -3, -1, 0.5]] 會移除

序列中第二個與第四個指數是非整數的未定義元素。 

反序 

Reverse[List]: 反轉指定串列的順序 

序列 

Sequence[Expression, Variable i, Number a, Number b]: 可產生一物

件序列，使用給定的表達式以及索引值 i 的範圍從數值 a 到數值 b。  
例: L = Sequence[(2, i), i, 1, 5] 可得到一點序列，其 y 座標範圍從 
1 到 5。 

Sequence[Expression, Variable i, Number a, Number b, Number s]: 
可產生一物件序列，使用給定的表達式以及索引值 i 的範圍從數值 a 到數值 
b ，間距大小為 s。  
例: L = Sequence[(2, i), i, 1, 5, 0.5] 可得到一點序列，其 y 座標

範圍從 1 to 5 ，間距大小為 0.5。 
 
註: 因為參數 a 與 b 是可變動的，所以可以使用 數值滑桿 來控制參數。 

排序 

Sort[List]: 對串列中的數值、文字物件或點做排序。  
註: 點串列是依照 x 座標做排序。  
例:  
Sort[{3, 2, 1}] 可得到串列 {1, 2, 3}.  
Sort[{"pears", "apples", "figs"}] 可得到按照字母順序排列的新串

列。  
Sort[{(3, 2), (2, 5), (4, 1)}] 可得到 {(2, 5), (3, 2), (4, 1)}. 

總和 

Sum[List]: 計算串列中所有元素之和。   
註: 此指令可用在數值、點、向量、文字、函數。 
例: 
Sum[{1, 2, 3}] 可得到數值 a = 6。  
Sum[{x^2, x^3}] 可得到 f(x)=x2 + x3。   
Sum[Sequence[i,i,1,100]] 可得到數值 a = 5050。  
Sum[{(1, 2), (2, 3)}] 可得到點 A = (3, 5)。  
Sum[{(1, 2), 3}] 可得到點 B = (4, 2)。  
Sum[{"a","b","c"}] 可得到文字 "abc"。 

Sum[List, Number n of elements]: 計算串列中前 n 個元素之和。   
註: 此指令可用在數值、點、向量、文字、函數。 
例: Sum[{1, 2, 3, 4, 5, 6}, 4] 可得到數值 a = 10。 

擷取 

Take[List, Start position m, End position n]: 擷取串列中第 m 個到第 
n 個元素成為新的串列 

聯集 

Union[List 1, List 2]: 合併兩個串列並移除重複的元素。 

3.3.18.  幾何變換指令 

伸縮  

UK English: 放大  
 
Dilate[Point A, Number, Point S]: 點 A 對點 S 做指定倍率的伸縮 
Dilate[Line, Number, Point S]: 直線對點 S 做指定倍率的伸縮  
Dilate[Conic, Number, Point S]: 圓錐曲線對點 S 做指定倍率的伸縮 
Dilate[Polygon, Number, Point S]: 多邊形對點 S 做指定倍率的伸縮  

註: 新的頂點與線段也會產生。  
Dilate[Image, Number, Point S]: 圖片對點 S 做指定倍率的伸縮 
 
註: 亦參見工具  Dilate Object from Point  

鏡射  

Reflect[Point A, Point B]: 將點 A 對點 B 做鏡射 
Reflect[Line, Point]: 將直線對指定的點做鏡射 
Reflect[Conic, Point]: 將圓錐曲線對指定的點做鏡射 
Reflect[Polygon, Point]: 將多邊形對指定的點做鏡射  

註: 新的頂點與線段也會產生。 
Reflect[Image, Point]: 將圖片對指定的點做鏡射 
Reflect[Point, Line]: 將點對指定的直線做鏡射 
Reflect[Line g, Line h]: 將直線 g 對直線 h 做鏡射 
Reflect[Conic, Line]: 將圓錐曲線對指定的直線做鏡射 
Reflect[Polygon, Line]: 將多邊形對指定的直線做鏡射 

註: 新的頂點與線段也會產生。 
Reflect[Image, Line]: 將圖片對指定的直線做鏡射  
Reflect[Point, Circle]: 點對圓進行反演。 
 
註: 亦參見工具  Reflect Object about Point;  Reflect Object about Line;  

 Reflect Point about Circle 

旋轉  

Rotate[Point, Angle]: 將點對原點旋轉指定的角度  
Rotate[Vector, Angle]: 將向量對原點旋轉指定的角度  
Rotate[Line, Angle]: 將直線對原點旋轉指定的角度  
Rotate[Conic, Angle]: 將圓錐曲線對原點旋轉指定的角度  
Rotate[Polygon, Angle]: 將多邊形對原點旋轉指定的角度  

註: 新的頂點與線段也會產生。  

Rotate[Image, Angle]: 將圖片對原點旋轉指定的角度  
Rotate[Point A, Angle, Point B]: 將點 A 對點 B 旋轉指定的角度 
Rotate[Line, Angle, Point]: 將直線對指定的點旋轉指定的角度  
Rotate[Vector, Angle, Point]: 將向量對指定的點旋轉指定的角度  
Rotate[Conic, Angle, Point]: 將圓錐曲線對指定的點旋轉指定的角度 
Rotate[Polygon, Angle, Point]: 將多邊形對指定的點旋轉指定的角度  

註: 新的頂點與線段也會產生。  
Rotate[Image, Angle, Point]: 將圖片對指定的點旋轉指定的角度 
 
註: 亦參見工具  Rotate Object around Point by Angle 

平移  

Translate[Point, Vector ]: 將點沿著給定之向量做平移  
Translate[Line, Vector]: 將線沿著給定之向量做平移  
Translate[Conic, Vector]: 將圓錐曲線沿著給定之向量做平移  
Translate[Function, Vector]: 將函數沿著給定之向量做平移 
Translate[Polygon, Vector]: 將多邊形沿著給定之向量做平移  

註: 會產生新的頂點與線段。  
Translate[Image, Vector]: 將圖片沿著給定之向量做平移  
Translate[Vector, Point]: 將向量 v 平移到指定的點 
 
註: 亦參見工具  Translate Object by Vector 

3.3.19.  統計指令 

長條圖 

BarChart[Start value, End value, List of heights]: 建立給定區間的

長條圖，長條數取決於表列的長度，而長條的高度取決於表列的元素。 
例: BarChart[10, 20, {1,2,3,4,5} ]:得到五條特定高度且區間為[10, 
20]的長條圖。 

BarChart[Start value a, End value b, Expression, Variable k, 
From number c, To number d] :建立給定區間[a, b]的長條圖，計算長條

的高度以從 c 到 d 的變數 k 表示之。 
例: 如果 p = 0.1, q = 0.9, 且 n = 10 ， 那麼  
BarChart[ -0.5, n + 0.5, 
BinomialCoefficient[n,k]*p^k*q^(n-k), k, 0, n ]:得到一個區間

為[-0.5, n+0.5]的長條圖。長條的高度取決於給定式子的機率。 
BarChart[Start value a, End value b, Expression, Variable k, 

From number c, To number d, Step width s]: 建立給定區間[a, b]的
長條圖。計算長條的高度以從 c 到 k 且每段距離為 s 來表示之。 

BarChart[List of raw data, Width of bars]: 建立給定原始資料及長條寬

度的長條圖。  
例: BarChart[ {1,1,1,2,2,2,2,2,3,3,3,5,5,5,5}, 1] 

BarChart[List of data, List of frequencies]: 建立對應的次數的資料表

的長條圖  
註: 資料表列必須是一串不變的數量 
例:  
BarChart[{10,11,12,13,14}, {5,8,12,0,1}]  
BarChart[{5, 6, 7, 8, 9}, {1, 0, 12, 43, 3}]  
BarChart[{0.3, 0.4, 0.5, 0.6}, {12, 33, 13, 4}] 

BarChart[List of data , List of frequencies, Width of bars w]: 
使用資料表列和對應的次數的長條圖，長條的寛度是 w 
註: 資料表列必須是一串不變的數量 
例: 
BarChart[{10,11,12,13,14}, {5,8,12,0,1}, 0.5] 長條之間留有空

隙  
BarChart[{10,11,12,13,14}, {5,8,12,0,1}, 0] 產生直線圖 

Box Plot 盒狀圖 

BoxPlot[yOffset, yScale, List of raw data]: 建立給定原始資料的盒狀

圖且在座標系統中的垂直方向是由 y 偏移量所控制，高度受到 y 進位因子所影

響  
例: BoxPlot[0, 1, {2,2,3,4,5,5,6,7,7,8,8,8,9}] 

BoxPlot[yOffset, yScale, Start value, Q1, Median, Q3, End 
value]: 在區間[Start value, End value]建立給定統計資料的盒狀圖 

相關係數 

CorrelationCoefficient[List of x-coordinates, List of y-
coordinates]: 使用給定的 x 座標和 y 座標來計算相關係數的乘積 

CorrelationCoefficient[List of points]: 使用給定點的座標來計算相關係

數的乘積 

共變異數 

Covariance[List 1 of numbers , List 2 of numbers]: 計算兩串列的共

變異數 
Covariance[List of points]: 計算 x 座標與 y 座標的共變異數。 

最適直線或迴歸線 

FitLine[List of points]: 計算 xy 座標上的線性迴歸 

其他相關迴歸指令 

FitExp[List of points]: 計算指數迴歸曲線 
FitLineX[List of points]: 計算 xy 座標平面上的線性迴歸 
FitLog[List of points]: 計算 logarithmic 迴歸曲線 

FitLogistic[List of points]: 計算 a/(1+b x^(-kx))形式的迴歸曲線 
註: 第一個與做後一個資料點會設定為曲線的端點。該串列至少必須包含三個

點。 
FitPoly[List of points, Degree n of polynomial]: 計算 n 次多項式的

迴歸方程。 
FitPow[List of points]: 計算 a xb 形式的迴歸曲線   

註: 所有的點必須在座標系統的第一象限。 
FitSin[List of points]: 計算 a + b sin(cx+d)形式的迴歸曲線 

註: 此串列至少要有四個點。 

直方圖 

Histogram[List of class boundaries, List of heights]: 藉由給定之

高度來建立一直方圖。  
 

Histogram[List of class boundaries, List of raw data]: 使用原始資

料建立一直方圖。 
例: Histogram[{1, 2, 3, 4},{1.0, 1.1, 1.1, 1.2, 1.7, 2.2, 
2.5, 4.0}] 建立一個長方圖，包含三個然欄位，高度 5(第一欄)、高度 2(第
二欄)、高度 1(第三欄)。 

InverseNormal 反向常態函數 

InverseNormal[Mean, Standard deviation, Probability]: 計算函數 Φ-

1(P) * σ + μ，其中 Φ(x) -1 是 Φ(x) 機率密度函數佈於 N(0,1)的反函數。 
註: 從給定的機率算出在常態分佈曲線下方 x 的座標，回傳 x 座標。 

平均值相關指令 

Mean[List of numbers]: 計算串列元素的平均數 
MeanX[List of points]: 計算串列元素 x 座標的平均數 
MeanY[List of points]: 計算串列元素 y 座標的平均數 

中位數 

Median[List of numbers]: 得到串列元素的中位數。 

眾數 

Mode[List of numbers]: 得到串列元素中的眾數。  
例: 
Mode[{1,2,3,4}] 回傳空串列 {}  
Mode[{1,1,1,2,3,4}] 回傳串列 {1}  
Mode[{1,1,2,2,3,3,4}] 回傳串列 {1, 2, 3} 

Normal 常態函數 

Normal[Mean, Standard deviation, Variable value]: 計算函數 (Φ(x) - 
mean ) / (standard deviation) 其中 Φ (x)是佈於 N(0,1) 中的機率密度函數。

  
註: 回傳對一個給定 x 座標數值的機率(或在給定 x 座標右方，常態分佈曲線下

方的區域)。 

四分位數指令 

Q1[List of numbers]: 得到串列元素中的第一四分位數 
Q3[List of numbers]: 得到串列元素中的第三四分位數 

標準差 

SD[List of Numbers]: 計算串列中數值的標準差 

Sigma 指令 

SigmaXX[List of numbers]: 計算給定數值之平方和  
例: 可以使用 SigmaXX[list]/Length[list] - Mean[list]^2 算出串

列的變異數。 
SigmaXX[List of points]: 計算給定點串列之 x 座標的平方和 
SigmaXY[List of x-coordinates, List of y-coordinates]: 計算兩點串

列的 x 與 y 座標乘積的總和。 
SigmaXY[List of points]: 計算 x 座標與 y 座標乘積之和。   

例: 可以使用 SigmaXY[list]/Length[list] - MeanX[list] * 
MeanY[list]得到該串列的共變異數。 

SigmaYY[List of Points]: 計算給定點串列之 y 座標的平方和 

統計量指令 

Sxx[List of numbers, List of numbers]: 計算 Σ(x2) - Σ(x) × Σ(x)/n 
Sxx[List of points]: 計算 Σ(x2) - Σ(x) × Σ(x)/n 
Sxy[List of numbers, List of numbers]: 計算 Σ(xy) - Σ(x) × Σ(y)/n 
Sxy[List of points]: 計算 Σ(xy) - Σ(x) × Σ(y)/n. 
Syy[List of numbers, List of numbers]: 計算 Σ(y2) - Σ(y) × Σ(y)/n 
Syy[List of points]: 計算 Σ(y2) - Σ(y) × Σ(y)/n. 
 
註: 這些統計量是非常態形式的 XY 的變異數及共變異數。Sxx = N var(X), Syy = N 
var(Y), and Sxy = N cov(X,Y)  
範例: 可以使用 Sxy[list] / sqrt(Sxx[list] Syy[list])來算出點串列中的

共變異係數。 

變異數 

Variance[List of numbers]: 計算串列元素的變異數 

3.3.20.  試算表指令 

CellRange 

CellRange[Start cell, End cell]: 產生指定格範圍的串列  
例: CellRange[A1, A3] 可得到串列 {A1, A2, A3} 

欄值 

Column[Spreadsheet cell]: 回傳非空格的欄值(從 1 開始)  
例: 如果 B3 格非空，則 Column[B3] 可得到數值 a = 2 ，因為 B 欄是試算表

中的第二欄。 

欄名 

ColumnName[Spreadsheet cell]: 回傳非空格的欄名當作文字  
例: 如果 A1 格是非空，則 ColumnName[A1] 可在幾何視區得到文字 “A”。 

列值 

Row[Spreadsheet cell]: 回傳非空格的列值 (從 1 開始)  
例: 如果 B3 格是非空，則 Row[B3] 可得到數值 a = 3. 

3.3.21.  矩陣指令 

行列式 

Determinant[Matrix]: 回傳方陣的行列式值  
範例: Determinant[{{1, 2}, {3, 4}}] 會回傳一個數值 c. 

反矩陣 

Invert[Matrix]: 回傳給定方陣之反矩陣  
範例: Invert[{{1, 2}, {3, 4}}] 會回傳反矩陣 
{{-2, 1}, {1.5, -0.5}}. 

轉置矩陣 

Transpose[Matrix]: 將給定的矩陣做轉置。  
例: Transpose[{{1, 2}, {3, 4}}] 可以得到轉置矩陣   
{{1, 3}, {2, 4}}. 

4.  選單 

4.1.  檔案選單 

 開新視窗  

快捷鍵: Ctrl-N (MacOS: Cmd-N) 
 
此選項開啟新的 GeoGebra 視窗，使用預設的 GeoGebra 介面設定。 
註: 若更改與 儲存 設定值，新的 GeoGebra 視窗將會依據自訂設定開啟。 

開新檔案  

此選項將在同一個 GeoGebra 視窗開啟新的空白介面。在開啟新檔案之前電腦會詢問

是否儲存先前的檔案。 
註: 新的使用者介面將會採取先前的設定。譬如，在開新檔案之前，假如座標軸被隱藏

起來，則再開新檔案時座標軸也將會被隱藏。 

 開啟…  

快捷鍵: Ctrl-O (MacOS: Cmd-O) 
 
此選項可以開啟電腦中 GeoGebra 檔案 (附加檔名為.ggb)。 
註: 可以拖曳 GeoGebra 檔案至 Geobebra 視窗，即可開啟 GeoGebra 檔案。 

 存檔  

快捷鍵: Ctrl-S (MacOS: Cmd-S) 
 
此選項可以將檔案儲存為 GeoGebra 檔案(副檔名為 .ggb)。 
註: 假如此檔案已經儲存了，存檔將會以相同的檔案名稱覆蓋先前的檔案。 

另存新檔… 

此選項可以將檔案另存為新的 GeoGebra 檔案(副檔名為 .ggb)。電腦將會要求輸入新

的檔名。 

 預覽列印 

快捷鍵: Ctrl-P (MacOS: Cmd-P) 
 
此選項將可以開啟幾何視區的 預覽列印視窗。可以選擇是否顯示‘標題’, ‘作者’, ‘日期’ 
與 ‘比例單位’(cm)。 

註: 欲改預覽列印的設定，在輸入完畢後按下 Enter 鍵即可更新。  

輸出 -  匯出網頁 (html)…  

快捷鍵: Ctrl-Shift-W (MacOS: Cmd-Shift-W) 
 
此選項可將檔案輸出為網頁，建立‘動態工作單’, ‘Applet’, ‘Mathlet’,…請參考 建立互動

網頁 章節，將提供更多訊息。 

輸出 -  匯出圖檔 (png, eps)…  

快捷鍵: Ctrl-Shift-P (MacOS: Cmd-Shift-P) 
 
此選項可將 GeoGebra 幾何視區儲存為 圖檔。在對話框可以選擇圖檔的格式與比例單

位(cm)以及解析度(dpi)。  
 
幾何視區輸出為圖檔可以選擇下面幾種格式：  

 PNG – Portable Network Graphics: 為一種畫素的圖形格式，解析度愈高 
(dpi), 品質愈好 (300dpi 一般就夠好了)， 避免對 PNG 圖形作縮放而降低影像

品質。PNG 圖形檔案適合用於網頁(html)及 Word 文件。 
 
註: 無論是否在 Word 文件中使用 PNG 圖形檔 (檔案 中的 圖形，插入 選單) 請
確認大小設定為 100 %，否則給定的大小(以 cm)將被改變。 
  

 EPS – Encapsulated Postscript: 為一個向量圖形格式。EPS 圖片縮放後可保

持原有品質。EPS 圖形檔很適合用在向量圖形程式例如: Corel Draw, 和專業的

文字處理軟體如 LaTeX。EPS 圖形的解析度都是 72dpi，此數值只用在以公分

計算圖形的真實大小，且不影響圖形的品質。 
 
註: 使用 EPS 在多邊形和圓錐曲線區內質地的透明度是無效的。 

 PDF – Portable Document Format (參考上述 EPS 格式 )  
 SVG – Scaleable Vector Graphic (參考上述 EPS 格式 ) 
 EMF – Enhanced Meta Format (參考上述 EPS 格式) 

輸出-  複製到剪貼簿  

快捷鍵: Ctrl-Shift-C (MacOS: Cmd-Shift-C) 
 
此選項可以複製幾何視區至電腦的剪貼簿。在這之後，可以剪貼圖片到其他文件(例
如：Word 文件)。 

輸出 - 匯出 PSTricks 檔…  

快捷鍵: Ctrl-Shift-T (MacOS: Cmd-Shift-T) 
 
此選項可將幾何視區儲存為 PSTricks 圖檔，為 LaTeX 圖片格式。 

輸出 – 匯出 PGF/TikZ 檔… 

此選項可將幾何視區儲存為 PGF/TikZ 圖檔，為 LaTeX 圖片格式。 

 關閉  

快捷鍵: Alt-F4 (MacOS: Cmd-W) 
 
此選項可關閉 GeoGebra 視窗，電腦將會詢問是否存檔。 

4.2.  編輯選單 

 復原 

快捷鍵: Ctrl-Z (MacOS: Cmd-Z) 
 
此選項將回到上一步的動作。 

註: 也可以使用工具列右邊的 復原按鈕。 

 重作 

快捷鍵: Ctrl-Y (MacOS: Cmd-Shift-Z) 
 
此選項將重作下一步的動作。 

註: 也可以使用工具列右邊的 重作按鈕。 

 複製到剪貼簿  

快捷鍵: Ctrl-Shift-C (MacOS: Cmd-Shift-C) 
 
此選項可以將幾何視區複製到電腦的剪貼簿。然後，可以複製剪貼圖片到其他文件(例
如：Word 文件)。 
 

 刪除  

快捷鍵: Delete 鍵 
 
此選項可將選取之物件與相依之物件刪除。 
註: 首先必須先選取欲刪除之物件。 

選擇全部  

快捷鍵: Ctrl-A (MacOS: Cmd-A) 
 
此選項將選取所有建構的物件。 

選擇目前圖層 

快捷鍵: Ctrl-L (MacOS: Cmd-L) 
 
此選項可選取在同一圖層的物件。 
註: 在此選項之前，必須先選取一個物件，作為圖層的依據。 

選取子物件  

快捷鍵: Ctrl-Shift-Q (MacOS: Cmd-Shift-Q) 
 
此選項可選取所有與該物件相依的子物件。. 
註: 在此選項之前，必須選擇‘母’物件。 

選取母物件 

快捷鍵: Ctrl-Q 
 
此選項可以選取所有物件的祖先，指物件都是依據該物件所建構的。 
註: 在此選項之前，必須選取這些相依的物件。 

 屬性… 

快捷鍵: Ctrl-E (MacOS: Cmd-E) 
 
此選項將開啟屬性對話框 ，其可以修改物件的屬性。  

4.3.  檢視選單  

 座標軸 

此選項將可顯示或隱藏幾何視區中的座標軸。 
註: 可以透過 幾何視區的屬性對話框來調整座標軸。 

 格線 

此選項可以顯示或隱藏幾何視區中座標的格線。 
註: 可以透過幾何視區的屬性對話框窗 來調整座標格線。 

代數視區 

快捷鍵: Ctrl-Shift-A (MacOS: Cmd-Shift-A) 
 
此選項可以顯示或隱藏代數視區。 

試算表視區 

快捷鍵: Ctrl-Shift-S (MacOS: Cmd-Shift-S) 
 
此選項可以顯示或隱藏試算表。 

輔助物件 

此選項可以顯示或隱藏代數視區中的輔助物件。 

窗格左右並排 

此選項可將 GeoGebra 不同視區以水平方式排列，若取消則以鉛直排列。 

指令列 

此選項可以顯示或隱藏指令列，此指令列在 GeoGebra 視窗的底部。 

指令選單  

此選項可以顯示或隱藏指令列中的指令選單。 

 構圖案本… 

此選項將在新視窗開啟構圖案本。 

導播欄 

此選項可以顯示或隱藏幾何視區中的導播欄。 

 清除所有蹤跡 

快捷鍵: Ctrl-F (MacOS: Cmd-F) 
 
此選項將重新整理螢幕上的所有視區。 
註: 可以使用此選項來刪除在幾何視區中點或線的蹤跡。  

重新計算所有物件 

快捷鍵: F9 
 
此選項將重新計算 GeoGebra 檔案中所有物件。 
註: 可以使用此選項來產生新的亂數。 

4.4 選項選單 

在選項選單中修改全域選項。 

註: 欲改變物件設定，請使用內容選單與屬性對話框。 

 吸附格線  

此選項可決定點被吸附於格子點的功能為開 / 關。 
註: 當座標系統的格子點顯現時，選擇“自動”將開啓吸附格線功能。如果格子點消失

則關閉此功能。  

 角的單位 

此選項可決定角度是否以角度 (°) 或徑度 (rad) 顯示。  
註: 輸入時兩種方式皆可(度數或徑度)。 

小數位數 

此選項可調整數或帶有標籤圖形的小數位數。  

連續性 

GeoGebra 可在選項選單中，可選擇 開/關 連續性。此軟體使用 near-to-heuristic 來保

持移動中的交點 (例如：直線與圓錐曲線, 圓錐曲線與圓錐曲線) 緊鄰舊的位置，來避

免交點跳動的情形。  
 
註: 預設「連續性」為關閉，使用者自訂工具 (參見使用者自訂工具) 連續性也都是關

閉狀態。 

 點的樣式 

此選項可決定點以● ，○，或 x 來顯示。  

 勾選框大小 

此選項可設定勾選框的大小 
註: 若使用 GeoGebra 作為呈現工具或互動式白板，設定勾選框的尺寸為 “大”時，

可更容易使用勾選框 。 

 直角樣式  

此選項可決定直角是否以矩形、點、或一般的角來顯示。 

座標 

此選項可決定點座標是否以 A = (x, y) 或 A(x | y) 的形式表示。. 

 物件標籤 

此選項可指定新增物件的標籤是否顯示。可以選擇開始顯示、開始不顯示、只顯示新

點標籤和自動。 

註:當開啟代數視區並新增物件時，自動設定為可顯示標籤。 

 字體大小  

此選項可決定標籤字體大小及文字(pt)。 
註: 若使用 GeoGebra 作為展示工具，增加字體大小有助於觀眾更容易閱讀文字、標

籤及正在使用的代數指令列。  

 語言 

GeoGebra 可使用多種語言，可在此更改現有的語言設定。如此會改變所有的輸入和

輸出的指令名稱。 
註:無論選擇哪種語言，此點擊地球圖案 會引導回到語言選單。所有的語言名稱都以

英文呈現。  

 繪圖區 

此選項可開啟幾何視區的屬性對話框(例如：座標軸、格線與背景顏色)。 
註: 亦可在幾何視區使用右鍵打開對話框(MacOS: Ctrl-click) 

 儲存設定  

若在選項選單中選取儲存設定，GeoGebra 會記憶(在選項選單、目前工具列和幾何視

區的設定)的偏好設定。 

回復預設值 

此選項可回復 GeoGebra 選單的預設值。  

4.5 工具選單 

 新增工具 

在 GeoGebra 中，可以依據現有的構圖，建立自訂工具。在準備好工具的建構後，在

工具選單中選擇新增工具。出現的對談框中，可指定工具的輸入物件及輸出物件，並

選擇工具列之圖示的名稱與指令。 
註: 新增的工具可使用滑鼠或在指令列中以指令輸入，所有工具都自動存成“ggb” 建構

檔。 

 管理自訂工具 

使用選取工具選單中的管理自訂工具時，可以刪除工具或修改名稱和圖像。也可以將

選取的工具儲存成 “ggt”格式的檔案。此檔案可以在其他的構圖上去讀取此工具(檔
案，開啓)。  
註: 開啓 “ggt”格式的檔案不會更改目前的構圖，但開啓 “ggb”檔案則會更改構圖。 

自訂工具列 

在工具選單中選取自訂工具列，即可在 GeoGebra 的工具列中自訂工具列。這對動態

工作單 特別好用，可以限制一些工具出現在工具列中。  
註: 目前的工具列設定將跟隨構圖一起儲存為“ggb” 的檔案。 

4.6 視窗選單 

 開新視窗  

快捷鍵: Ctrl+N  
 
此選項開啓預設的 GeoGebra 新視窗。  
註: 如果改變且儲存某些設定，那麼 GeoGebra 的新視窗將依照自訂的設定來開啓。 

開啓 GeoGebra 視窗表 

若有一個以上的 GeoGebra 視窗，則可使用此選項切換不同的視窗。 
註: 使用 GeoGebra 為呈現工具且想要擁有數個 GeoGebra 檔案同時開啓時，此選項

是很有用的。 

4.7 說明選單 

 說明 

此選項可進入線上的 GeoGebra 說明文件，且依據在電腦上安裝 GeoGebra 的方式，

可能需要網際網路才能開啓此文件。 
 若是從網頁下載和安裝 GeoGebra 的執行檔，使用此文件時不需要連上網際網

路，線上 HTML 版本的 GeoGebra 說明文件已儲存於電腦。 
 若是使用 GeoGebraWebstart 來安裝 GeoGebra，需要連上網際網路才能開啓

此文件，否則會得到錯誤訊息。 
 
註: 線上 HTML-版本的 GeoGebra 說明文件於 http://www.geogebra.org/help。 

 www.geogebra.org 

若連上網際網路，此選項可開啓 GeoGebra 的網頁 (http://www.geogebra.org)。 

 GeoGebra Forum 

若連上網際網路，此選項會開啓 GeoGebra 的使用者線上討論區

(http://www.geogebra.org/forum)。 
 
註: 與 GeoGebra 有關的問題可以在此討論區發表和回覆。 

 GeogebraWiki 

若連上網際網路，此選項會開啓 GeoGebra Wiki (http://www.geogebra.org/wiki)。 
 
註: GeoGebra Wiki 是來自世界各地的使用者創造教學素材的地方。  

 關於/版權 

此選項開啓對話視窗提供有關於 GeoGebra 版權的訊息和給予以不同方式貢獻且支持

GeoGebra 計劃的人們讚揚(例如：程式設計、翻譯)。 

5.  GeoGebra 特殊功能 

5.1.  動畫 

自動動畫 

GeoGebra 可以允許一個或數個自變數值或角度同時進行動畫，而這些動畫在幾何視

區中會使用滑桿來顯示。  
 
在 GeoGebra 中，若要使得自變數值或角度進行動畫，必須在該數值或角度上點擊滑

鼠右鍵 (MacOS: Ctrl＋滑鼠左鍵)，然後從內容選單中選擇動畫啟動(Animation On)。
若要停止動畫，相同的從內容選單中取消選取動畫啟動。 
 
註：在自變數字或自變角度啟動動畫後，在幾何視窗的左下角會出現動畫按鈕。此按

鈕可以 暫停或 繼續動畫。 
 
在屬性對話框中滑桿中，可以更改動畫的行為： 
 
一方面可以控制動畫的速度。 
註：速度設定 1，表示經過滑桿的一個區間需要 10 秒。 
 
另外一方面可以更改動畫循環的方式： 

 <=> 來回反覆:  
動畫循環會不停地來回反覆。 

 => 遞增：  
滑桿數值永遠會以遞增的方式變動。當它達到最大值時，就會跳回最小值，

然後繼續動畫。  
 <= 遞減：  

滑桿數值永遠會以遞減的方式變動，當它達到最小值時，就會跳回最大值，

然後繼續動畫。 
 
註：當啟動自動動畫的功能時， GeoGebra 仍然可使用其它的功能。此功能可使在動

畫播放過程中，同時可以更改構圖。  

手動動畫  

以手動的方式來改變一個數值的話，先選則  移動  工具，然後點擊該自變數字或角

度，然後再按鍵盤上的+或-鍵、或鍵盤的上下鍵。持續按壓這些鍵可產生手動動畫。  
 
範例：若一個點座標隨著數值 k 而變，譬如，P = (2 k, k)。當 k 在連續地更改時，此

點會沿著直線移動。  

 註： 在物件的屬性對話框中，在滑桿中可以調整滑桿的增量。  
 
快捷鍵：  

 「Shift + 方向鍵」可以一次跳 0.1 單位 
 「Ctrl + 方向鍵」可以一次跳 10 單位 
 「Alt + 方向鍵」可以一次跳 100 單位 

 
 註：直線上的一點也可以使用「+」或「-」鍵來移動。  

5.2.  條件顯示 

除了可以直接顯示或隱藏某些物件之外，亦也可以設定一個物件的可視狀態(何時顯

現、何時隱藏)的條件。例如，可以在「幾何視區」中新增一個勾選框，當這個勾選框

被勾選的時候，某個物件才會顯現；或者可以設定當滑桿改變至某特定值時，該物件

才會顯現。 

有條件地顯示現有物件 

可以使用  顯示或隱藏群組物件之選項 工具來新增一個勾選框，並利用這個勾選框來

顯示一個或多個物件。另外，也可使用指令列建立一個布林變數 （例如： b = 
true) ，然後藉由此變數的改變來幾何視區中物件的可視狀態(例如：使用工具  顯
示或隱藏群組物件之選項或使用內容選單)。為了使用布林變數當作某一個物件的可視

條件，必須依據下例步驟。 

改變新建立物件的顯示條件 

在物件的屬性對話框之進階，可以設定物件的顯示條件。 
  
註： 當在設定物件的顯示條件的時候，可以使用右邊下拉式選單中的邏輯運算符號

（例如：≠、 ≥、 ˄、 || 等等） 
 
範例： 

 若 a 是數值滑桿，然後在顯示條件中輸入 a < 2，那表示當數值 a 小於 2 的時

候，該物件才會顯現在幾何視區。 
 若 b 是布林變數，可使用 b 為條件敘述。當 b 的值為 true 時，該物件才會顯

現；當 b 的值為 false 時，該物件會隱藏起來。  
 若 g 與 h 是兩條直線，可以使用 g‖h 為條件敘述，表示當兩條線平行的時候，

該物件才會顯現。 

5.3.  自訂工具 

GeoGebra 可以建立自己的構圖工具。一旦工具建立，自訂的工具同時可以使用滑鼠

或在指令列中輸入指令。所有的工具會自動地儲存在 GeoGebra 檔案。 

建立自訂工具  

首先，必須先建立工具的構圖，然後在工具選單中點擊選項  新增工具來開啟相關的

對話框。必須在輸出物件頁、輸入物件頁、以及名稱與圖示頁中填入相關的資訊，以

建立自訂工具。 
 
範例：  
建立一個可以產生正方形的工具。在幾何視區點擊兩個點或點擊空白處，它就會幫建

立一個正方形。 
 建構以兩個自由點 A 與 B 為起點的正方形。先建立其他的頂點，然後使用工具

 多邊形連接各個點來得到一個正方形(poly1)。  
 選取工具選單中的新增工具。 
 指定輸出物件：用滑鼠點擊該正方形或從下拉式選單中選取。 也要指定正方形

的邊為輸出物件。 
 指定輸入物件：GeoGebra 會自動指定輸入物件 (這裡為點 A 與點 B)。若要更

改輸入物件，亦可使用下拉式選單或在構圖上點選。 
 指定工具名稱與指令名稱 

註：指定的工具名稱會出現在工具列中。指令名稱可以在 GeoGebra 的指令列

中使用。 
 填寫工具說明：若填寫工具說明的話，它會出現在工具列說明中。 
 指定工具圖示：可以從電腦中選擇圖片，當作工具圖示，GeoGebra 會自動調

整圖檔縮到適和工具列按鈕的大小。 

儲存自訂工具  

可以儲存自訂的工具，然後在其他的 GeoGebra 構圖中使用。首先，在工具選單，選

取  管理工具，然後選取從出現的選單中取選要儲存的自訂工具。點選另存新檔…按

鈕來儲存工具。 
註： 工具檔的副檔名為「.ggt」，一般的 GeoGebra 檔副檔名為「.ggb」。 

使用自訂工具  

建立了自訂工具後，從檔案選單中選擇開新檔案，開啟新的 GeoGeobra 介面，自訂

工具還會在工具列，但是，改天開啟了一個新的 GeoGebra 視窗，或 GeoGebra 檔

案，若自訂工具已經不在工具列中。  
 
下列幾種不同的方法，可再新的 GeoGebra 視窗的工具列顯示自訂工具： 

 在建立新的自訂工具後，可以從選項選單中使用儲存設定。此後，自訂工具會

一直顯示在工具列中。 

 註：若要從工具列中移除自訂工具，可從工具選單中開啟自訂工具列…。然

後，在出現的對談視窗中，左邊的工具表列中選取將移除的工具，最後點擊移

除＞按鈕。在移除工具後，不要忘了要儲存設定。 
 

 若已經儲存自訂工具（為 .ggt 檔），可以隨時將此檔案匯入新的 GeoGebra 視

窗。可以從檔案選單中的選項開啟來打自訂工具的檔案。 
註：開啟工具檔並不會將影響既有的構圖，自訂工具只會出現在工具列。 

5.4.  動態色彩 

在 GeoGebra 中，可以使用物件的屬性對話框中的顏色，可以更改物件的顏色。然

而，也可以讓此物件的顏色以動態地更改：開啟物件的屬性對話框在進階中，就會發

現一個動態色彩，裡面有「紅、綠、藍」三個文字欄位。 
 註:在這些文字欄位中，可以輸入數值介於 0 與 1 之間的任何函數。 
 
範例： 

 建立三個數值滑桿  a, b, c ，並讓它們的範圍介於 0 與 1 之間。 
 建立一個多邊形，讓它的顏色可以由數值滑桿的數值影響。 
 開啟此多邊形的屬性對話框，在進階中，然後在三個欄位中分別輸入 a, b, c 。 
 關閉屬性對話框，更改數值滑桿的數值，可以發現多邊形的顏色會隨著不同數

值產生變化。 
 註：也可以讓這些數值以不同的速度讓滑桿產進行動畫，可看到多邊形顏色自

動地改變。 

5.5.  JavaScript 介面 

註：GeoGebra 的 JavaScript 介面對於有使用 HTML 編輯網頁碼經驗的人而言，是一

個有趣的功能。 
 
為了讓動態工作單擁有更強的互動性，GeoGebra 提供了一個 JavaScript 介面。例

如，可以在網頁中建立一個按鈕，來產生動態構圖的圖像。  
 
請參考 GeoGebra Applets and JavaScript 文件(http://www.geogebra.org)，這裡有許

多例子與訊息關於使用 JavaScript。 

5.6.  快捷鍵 

鍵盤 無組合鍵 Ctrl 
(MacOS: Cmd)

Ctrl-Shift 
(MacOS: 

Cmd-Shift) 

Alt 
(MacOS: Ctrl) 

鍵盤 無組合鍵 Ctrl 
(MacOS: Cmd)

Ctrl-Shift 
(MacOS: 

Cmd-Shift) 

Alt 
(MacOS: Ctrl) 

A  全選 
顯示／隱藏  
代數視區  

符號 α 

B    符號 β 

C  貼上(試算表) 
匯出幾何視區 
至剪貼簿 

 

D    符號 δ 

E  「屬性」視窗   符號 ℯ 
F  重新計算亂數  符號 φ 

G    符號 γ 
H     
I     
J     
K     
L  選取目前圖層  符號 λ 

M    符號 μ 

N  開新視窗   
O  開啟舊檔  角度符號 ° 

P  預覽列印 
匯出幾何視區 
為圖檔 

圓周率 π 

Q  選取子物件 選取母物件  
R     

S  儲存 
顯示／隱藏  
試算表  

符號 σ 

T   
匯出

PSTricks 檔 
符號 θ 

U     
V  貼上 (試算表)   

W  
關閉  
(MacOS 專用) 

匯出動態網頁  

X     
Y  重作   
Z  復原   
0    指數（次方） 0 

1    指數（次方）1 

2    指數（次方） 2 

3    指數（次方） 3 

4    指數（次方） 4 

5    指數（次方） 5 

6    指數（次方） 6 

7    指數（次方） 7 

8    指數（次方） 8 

9    指數（次方） 9 

- 遞減（數值） 縮小圖形  符號 ∓ 

+ 遞增（數值） 放大圖形  符號 ± 

= 遞增（數值） 放大圖形  符號 ≠ 

<    符號 ≤ 
, (逗號)    符號 ≤ 

>    符號 ≥ 
. (句號)    符號 ≥ 

鍵盤 無組合鍵 Ctrl 
(MacOS: Cmd)

Ctrl-Shift 
(MacOS: 

Cmd-Shift) 

Alt 
(MacOS: Ctrl) 

*    複數乘法 ⊗ 

F1 說明    

F2 
編輯所選物件 
(代數視區) 

   

F3 將游標移至指令列    
F4     
F9 重新計算亂數    

Enter 
在「幾何視區」與

「指令列」間切換 
   

滑鼠左鍵      

滑 鼠 右 鍵  
(MacOS:Ctrl-
click)  

按一下： 
打開「內容選單」

（物件） 
 
打開「幾何視區 」
的「屬性」（空白

處） 
 
拖曳： 
移動（物件） 
 
放大框選區域（空白

處） 

   

滑鼠滾輪 放大／縮小 
放大／縮小 
(動態網頁中) 

  

Delete 刪除    
Backspace 刪除    

向上鍵 ↑  

遞增（數值） 
 
向上移動（點） 
 
前一個指令 
（指令列） 
 
上一個步驟 
（構圖案本）  

10 倍速 
0.1 倍速 
(只按 Shift) 

100 倍速 

向右鍵 → 

遞增（數值） 
 
向右移動（點） 
 
上一個步驟 
（構圖案本） 

10 倍速 
0.1 倍速 
(只按 Shift) 

100 倍速 

向左鍵 ← 

遞減（數值） 
 
向左移動（點） 
 
下一個步驟 
（構圖案本） 

10 倍速 
0.1 倍速 
(只按 Shift) 

100 倍速 

鍵盤 無組合鍵 Ctrl 
(MacOS: Cmd)

Ctrl-Shift 
(MacOS: 

Cmd-Shift) 

Alt 
(MacOS: Ctrl) 

向下鍵 ↓ 

遞減（數值） 
 
向下移動（點） 
 
後一個指令 
（指令列） 
 
下一個步驟 
（構圖案本） 

10 倍速 
0.1 倍速 
(只按 Shift) 

100 倍速 

Home/PgUp 
在構圖案本中回到第

一個步驟。 
   

End/PgDn 
在構圖案本中跳到最

後一個步驟。 
   

 
其他快捷鍵： 

 Alt-Shift (MacOS: Ctrl-Shift)：希臘大寫字母 
 試算表視區：Ctrl-Alt-C 貼上數值（而不是公式） 

 
註： 角度符號「°」 (Alt-O, MacOS: Ctrl-O)和圓周率「π」 (Alt-P, MacOS: Ctrl-P) 也
可以用於設定數值對談框的「最大值、最小值」與「增量」等欄位。 

5.7.  物件名稱與標籤 

顯示與隱藏名稱 

在幾何視區有好幾個不同的方法可以顯示或隱藏物件的名稱： 
 選擇工具  顯示或隱藏名稱 ，然後點擊要顯示或隱藏名稱的物件。 
 開啟開物件的內容選單，然後選擇  顯示標籤。 
 開啟物件的屬性對話框，在一般中勾選或取消勾選顯示標籤。 

名稱與數值 

在 GeoGebra 中，每個物件都有唯一的名字來標示該物件的名稱。另外，物件也可以

使用它的數值或名字來標示。可以在物件的屬性對談框中的一般頁設定，以勾選只顯

示物件名稱、只顯示物件數值或名稱與數值一起顯示。 
 
 註:一個點的數值為一個座標，一個函數的數值為其方程式。 

標籤 

有時後可以給許多物件相同的名稱，譬如，標示正方形的邊為 a。在這個例子中，

GeoGeobra 對所有物件提供標籤，可以附加在物件上。可以在物件的屬性對話框中一

般中在標籤的文字方塊中輸入文字來設定物件的標籤。 

5.8.  圖層 

註:在 GeoGebra 中，當使用者同時點選多個物件時，圖層被使用來決定哪些物件會被

選取或拖曳。 
 
預設值，所有物件被繪製在圖層 0，也就是所謂的幾何視區的背景層。總共有 10 個 
圖層（0 到 9），越大的號字表示越上。 
 
使用屬性對話框的進階，可以更改特定物件的所屬圖層(可使用的圖層從 0 到 9)。一旦

更改圖層數字(譬如，改為圖層 3)，所有新的物件將會被繪製在最高圖層。 
 
 註：可以選取同一層的所有物件，在選取任意一個物件後，然後在編輯選單中選擇選

取目前圖層（或快捷鍵 Ctrl-L）。此選單選項選取同一圖層的所有物件。  
 
圖層的進階使用： 

 會出 SVG 格式檔案，物件會以圖層來分群。 
 圖層也可以使用 JavaScript 介面來控制。 

5.9.  重新定義 

當要更改構圖時，重新定義物件是一個非常靈活的工具。注意這些改變也可能會改變

在構圖按本中構圖步驟的排序。 
 
在 GeoGebra 中，有下列幾個不同的方式來重新定義物件： 

 在代數視區中，選取  移動工具，然後雙擊任何物件。 
o 若是自變物件，編輯區是開放的，可以允許直接修改物件的代數表徵，

修改完按 Enter 鍵即可。 
o 若是應變物件，從新定義對話框會開啟，允許從新定義物件。 

 在幾何視區中，選擇  移動工具，然後雙擊任意物件。就會出現一個重新定義
對話框，可以從新定義物件。 

 在指令列中，可以直接輸入該物件的名稱與新的定義，來更改物件 。 
 開啟物件屬性對話框的一般，然後更改物件的定義。 

 
 註:固定物件無法重新定義。若要重新定義固定物件，必須先到該物件屬性對談框的一

般，取消固定物件這個勾選框。 
 
範例： 

 若要讓一個自由點 A 放置在直線 h 上的話，首先雙擊該 A 點開啟從新定義對話

視窗。然後在出現的文字方塊中輸入指令 Point[h]在按下 Enter 鍵。若要讓

該點離開線上，變回自由點，必須將它重新定義成一個自由的座標，譬如(1, 
2)。  

 若要讓一條通過 A 與 B 兩點的直線，變成 AB 線段。開啟物件直線 h 的重新定

義對話框，然後在出現的文字方塊中修改指令為 Segment[A, B]即可。 

5.10.  蹤跡與軌跡 

在幾何視區  中，當物件移動的時候，可以讓它留下蹤跡。使用內容選單並切換  顯
示移動蹤跡。然後，當修改構圖時，會追蹤物件位置的變動，然後留下蹤跡。 
 
註 ：若要關閉物件的蹤跡，可以在內容選單中取消勾選  顯示移動蹤跡。在檢視選單

中  清除所有蹤跡會清除所有的蹤跡。  
 
另外，可以讓 GeoGebra 自動建立點的軌跡。可以使用滑鼠選擇  軌跡工具，或在指

令列中以直接輸入 Locus 指令。 
 
註 ：若要建立動點的軌跡，則此點必須依據另一點的移動，可能沿著某個物件移動

（如：直線、線段、圓等）。 
 
範例： 

 建立一個線段於 A = (-1, -1)、B = (1, -1)兩點之間。 
 在線段上取一點 C，這時點 C 只能沿著線段移動。 
 建立一點 P，是與 C 點相關（譬如，P = (x(C), x(C)^2)。 
 使用工具或指令來建立 P 點的軌跡： 

o  軌跡工具：先點擊點 P，然後再點擊點 C。 
o 指令 Locus：在指令列中輸入 Locus[P, C]，然後按 Enter 鍵。 

註： 在範例中會建立一個 [-1, 1] 區間的拋物線。 

